
Chapitre 8  –  Solutions des problèmes 

 

 

 

1. Le congrès de l’Association internationale de marketing. 

 

(a) La figure ci-dessous indique comment nous avons procédé.  Tout d'abord, la formule =ALEA() 

de la colonne B commande à Excel de générer 1000 nombres pseudo-aléatoires entre 0 et 1.  Ces 

derniers sont ensuite convertis en valeurs pseudo-aléatoires de la variable X; les paramètres NMoy 

et NÉcartT de la formule en C5 représentent la valeur moyenne 4500 et l'écart type 250 de la 

variable normale X.  Enfin, les colonnes D à M donnent les coûts qu'aurait encourus l'Association 

si le nombre de réservations avait été fixé à 4350, 4370, 4390, etc.  La formule de la plage 

D5:M1004 couvre deux cas d'espèce. 

 Considérons d'abord la cellule D5.  Le nombre x = 4517 de congressistes (voir cellule C5) est 

supérieur au nombre r = 4350 de réservations (D4).  L'Association doit payer 3 nuitées pour 

chacune des chambres supplémentaires requises par les x – r = 167 congressistes en 

excédent. Le coût à reporter en D5 s'élève donc à  15 030 dollars : 

3 × (80 – 50) × (x – r) = 3 × 30 × 167 = 15 030. 

De même, on retrouve en E5 un coût de 13 230 dollars : 

3 × (80 – 50) × (x – r) = 3 × 30 × (4517 – 4370) = 3 × 30 × 147 = 13 230. 

 Par contre, dans le cas traité en M5, le nombre x = 4517 de congressistes est inférieur au 

nombre r = 4530 de réservations (M4).  L'Association doit payer les r – x = 13 chambres en 

trop, soit 1 950 dollars au total : 

3 × 50 × (r – x) = 3 × 50 × (4530 – 4517) = 3 × 50 × 13 = 1 950. 

 

Fichier : MOG8-01.xlsx Feuilles : SimulA et  Fig1 

 
Cellule Formule Copiée dans :  

B5 =ALEA() B6:B1004 

C5 =ARRONDI(LOI.NORMALE.INVERSE.N(B4;NMoy;NÉcartT);0) C6:C1004 

D5 =3*SI(D$4>=$C5;50*(D$4-$C5);30*($C5-D$4)) D5:M1004 



2   Chapitre 8  –  La simulation 

La figure suivante résume les résultats obtenus. On constate (voir ligne 8) que la meilleure 

stratégie semble de réserver environ 4410 chambres. 

 

Fichier : MOG8-01.xlsx Feuilles : RésultatsA et  Fig2 

 
Cellule Formule Copiée dans :  

B5 =MIN(SimulA!D$5:D$10044) C5:K5 

B6 =MEDIANE(SimulA!D$5:D$1004) C6:K6 

B7 =MAX(SimulA!D$5:D$1004) C7:K7 

B8 =MOYENNE(SimulA!D$5:D$1004) C8:K8 

B9 =B8-1,645*ECARTYPE(SimulA!D$5:D$1004)/1000^0,5 C9:K9 

B10 =B8+1,645*ECARTYPE(SimulA!D$5:D$1004)/1000^0,5 C10:K10 

 

(b) On reprend les calculs de la feuille SimulA, mais en reportant en C5 la formule suivante : 

=ARRONDI(SI(B5<=Tk;TMin+(TMax-TMin)*RACINE(Tk*B5);TMax-(TMax-TMin)*RACINE((1-Tk)*(1-B5)));0) 

Les paramètres TMin et TMax représentent les valeurs minimale et minimale de la variable 

triangulaire X.  Enfin, Tk correspond à la proportion de l'intervalle [TMin; TMax] située à la gauche 

de la valeur modale TMod et se calcule selon la formule donnée à la page 441 : 

𝑇𝑘 =  
𝑇𝑀𝑜𝑑−𝑇𝑀𝑖𝑛

𝑇𝑀𝑎𝑥−𝑇𝑀𝑖𝑛
 =  

4000−2300

7000−2300
 =  0,362. 

 

La formule ci-dessus s'obtient de la formule correspondante de la question (a) en remplaçant la 

fonction LOI.NORMALE.INVERSE.N d'Excel par la fonction inverse d'une loi triangulaire (voir la 

formule de la cellule G6 de la figure 8.19). 

 

La figure suivante résume les résultats obtenus. On constate (voir ligne 8) que la meilleure 

stratégie semble de réserver entre 4010 et 4030 chambres. 

 

Fichier : MOG8-01.xlsx Feuilles : RésultatsB et Fig3 

Note 1.  Dans la figure 1 ci-dessus, nous avons utilisé les mêmes nombres pseudo-aléatoires pour 

comparer les différentes stratégies qui s'offrent à l'Association : en effet, les coûts calculés dans les 

colonnes D à M sont tous basés sur les valeurs de la variable X qui se trouvent dans la colonne C.  Nous 

avons procédé de la même façon dans la question (b).  Cette approche découle de la règle, énoncée à la 

page 419, qui recommande d'apparier, autant que possible, les données employées pour comparer des 

groupes.  Cependant, les analyses effectuées en (a) et en (b) sont indépendantes, et il ne nous a pas semblé 

nécessaire de reprendre en (b) les mêmes nombres pseudo-aléatoires qu'en (a). 

 

Note 2.  Il était difficile a priori de deviner la valeur approximative de nombre optimal de réservations. 

Dans la question (a), nous avons effectué un premier essai en comparant les coûts lorsque le nombre r de 

réservations variait de 4000 à 4900 par pas de 100.  Nous avons alors constaté que le coût minimal 

semblait être obtenu quand r prenait une valeur près de 4400.  Nous avons donc raffiné notre analyse en 

procédant à un deuxième essai dans lequel r variait de 4350 à 4450 par pas de 10.  De même, en (b), nous 

avons d'abord fait varier r de 3000 à 4800 par pas de 200. 

 

 

2. Une chaîne de montage en nanotechnologie. 

 

La figure ci-dessous indique comment nous avons procédé.  Nous commenterons d'abord les 

formules des colonnes B à E qui servent à répondre à la question (a).  Nous expliquerons celle de 

la dernière colonne dans la réponse à la question (b). 

 

Fichier : MOG8-02.xlsx Feuilles : Simul et Fig1 

 

 

 

 

 

 

 

 

 

 

 
 

Cellule Formule Copiée dans :  

B2 =AMin+ARRONDI(ALEA()*(AMax-AMin+1)-0,5;0) B3:B1001 

C2 =BMin+ARRONDI(ALEA()*(BMax-BMin+1)-0,5;0) C3:C1001 

D2 =CMin+ARRONDI(ALEA()*(CMax-CMin+1)-0,5;0) D3:D1001 

E2 =MIN(B2;C2;D2) E3:E1001 

F2 =RevU*E2-CtA*MAX(B2-E2;0)-CtB*MAX(C2-E2;0)-CtC*MAX(D2-E2;0) F3:F1001 

 

(a) Les formules des colonnes B à D génèrent des nombres pseudo-aléatoires obéissant à des lois 

uniformes discrètes.  Analysons plus en détail celle de B.  Les paramètres AMin = 45 et AMax = 65 

représentent les valeurs extrêmes de la variable discrète XA, le nombre de composantes produites 

une journée donnée par le laboratoire A. 

Observons d'abord que la variable discrète XA peut prendre 21 valeurs, soit 45, 46, 47, …, 65.  

Nous nous fixerons comme premier objectif de générer de façon équiprobable des nombres 

entiers pseudo-aléatoires entre 0 et 20.  Le terme ALEA() de la formule commande de générer des 

nombres réels entre 0 et 1.  En multipliant par (AMax – AMin + 1) = 21, puis en retranchant 0,5, on 

obtient un nombre réel entre –0,5 et 20,5, que l'on transforme ensuite en nombre entier à l'aide de 

la fonction ARRONDI d'Excel.  Noter que le terme –0,5 est nécessaire, sinon 22 valeurs seraient 

possibles, de 0 à 21 inclusivement (cette dernière valeur surviendrait lorsque la fonction ALEA() 

renvoie un nombre supérieur à 1 – 1 / (2×21) = 0,9762). 

 

Il suffit ensuite d'additionner la constante AMin pour obtenir le nombre de composantes produites 

pendant la journée par le laboratoire A, qui sera un nombre entier entre AMin = 45 et AMax = 65. 

 

Enfin, la formule de la colonne E est la définition de la variable Y. 

 

Le tableau ci-dessous résume les résultats  obtenus.  Les formules utilisées sont semblables à 

celles de la figure 2 du problème 1 et ne sont pas reproduites ici.  On constate que Y varie de 40 à 

65, avec la valeur 47 comme médiane.  De plus, le nombre moyen de nano-objets assemblés en 

une journée s'élève à 48,32 environ. 

 

Fichier : MOG8-02.xlsx Feuilles : Résultats et Fig2 

 

 

(b) Le revenu quotidien net de la chaîne de montage est calculé dans la colonne F de la figure 1.  

Les noms RevU, CtA, CtB et CtC font référence à des cellules de la feuille Données et représentent 

les paramètres de notre modèle : 

RevU = 3600   et   CtA = 875   et   CtB = 560   et   CtC = 1200. 

Le premier terme de la formule en F2 donne le profit brut de la journée : il s'agit tout simplement 

de multiplier le profit unitaire par le nombre de nano-objets assemblés.  De ce montant, il faut 

retrancher les coûts de mise au rebut.  Par exemple, pour les deux premières journées considérées 

à la figure 2, le revenu net s'élève à 172 850 $ et à 171 520 $ respectivement : 

 (3600 × 52) – (875 × 10) – (560 × 10) – (1200 × 0)  =  172 850 

 (3600 × 51) – (875 × 0) – (560 × 13) – (1200 × 4)  =  171 520. 

La colonne F de la figure 2 donne les principales caractéristiques de la distribution du revenu net 

pour les 1000 journées de l'échantillon.  On constate que la chaîne procure un profit quotidien 

moyen de 156 603 $, mais que le gain net varie considérablement d'une journée à l'autre, s'étalant 

d'un minimum de 105 955 $ à un maximum de 203 000 $.  L'histogramme ci-dessous décrit la 

distribution du revenu net (en k$) dans l'échantillon considéré aux figures 1 et 2. 

 

Note 1.  Nous indiquons brièvement comment nous avons obtenu cet histogramme. 

 Étape préliminaire (non essentielle) : nous avons créé une feuille Hist-RevN et nous avons copié dans 

la plage A2:A1001 les  revenus nets de la colonne F de la feuille Fig2; nous avons ensuite divisé ces 

nombres par 1000, afin d'alléger les étiquettes de l'axe horizontal.   

 Construction proprement dite du graphique : cliquer sur le menu Données, puis sur l'option Utilitaire 

d'analyse et enfin sur Histogramme. Pour l'exemple traité ici, la boîte Histogramme a été complétée 

de la façon suivante : 

o Plage d'entrée :  A2:A1001 

o Plage des classes :  D2:D14 

o Plage de sortie :  G1:L10 

o Cocher Représentation graphique. 

 Mise en forme fine du graphique (non essentielle) : nous avons modifié le titre du graphique, ainsi 

que celui de l'axe vertical; nous avons supprimé le titre de l'axe horizontal et reporté «230» en G15; 

pour afficher les effectifs des classes, nous avons coché droit sur l'un des rectangles et avons cliqué 

sur Ajouter des étiquettes de données; de même, pour que les classes soient contiguës, nous avons 

coché droit sur l'un des rectangles et nous avons cliqué sur Mettre en forme une série de données, 

puis sur Options des séries et enfin, dans Largeur de l'intervalle, nous avons déplacé le curseur à Pas 

d'intervalle. 

 

Note 2.  Il faut ici utiliser des nombres pseudo-aléatoires différents dans les colonnes B, C et D de la 

figure 1, car les quantités fabriquées par les trois laboratoires sont des variables aléatoires indépendantes. 

3. Les télescopes de Celistron. 

 

(a) La figure ci-dessous indique comment simuler la vente de 1000 appareils.  Tout d'abord, en 

colonne B, on génère 1000 nombres pseudo-aléatoires obéissant à une loi normale de moyenne 4 

et d'écart type 1,5 (les noms MoyenneX et EcartTX de la formule en B3 renvoient à des cellules 

de la feuille Données contenant les valeurs 4 et 1,5). 

 

La plage B3:B1002 donne les durées de vie des 1000 télescopes de notre échantillon.  Considérons 

le 1000
e
 et dernier cas de la figure (voir ligne 1002).  Cet appareil fonctionnera correctement 

pendant seulement 1,38 année, ce qui est inférieur aux 2 ans de la garantie, et Celistron devra le 

remplacer.  Le nouveau télescope aura lui aussi une durée de vie aléatoire obéissant à une loi 

normale N(4; 1,5
2
).  Il se trouve (voir cellule C1002) qu'il tombera en panne après 1,92 année. 

Celistron fournira donc au client un 3
e
 télescope qui durera 4,64 années, soit nettement plus que la 

garantie.  En E1002, on obtiendra donc une réponse positive quand on testera si le contenu de 

D1002 est égal ou supérieur à 2; par conséquent, Excel va renvoyer une cellule vide en E1002. 

 

La cellule C1003 compte combien de cellules ne sont pas vides dans la plage C3:C1002, c'est-à-

dire combien, parmi les 1000 télescopes vendus, devront être remplacés une première fois par 

Celistron.  De même, D1003 combien de télescopes de remplacement tomberont en panne 

prématurément et devront eux aussi être remplacés.  Enfin, la cellule B1003 indique combien 

d'appareils au total Celistron remettra gratuitement aux clients : 

69 + 10 + 3 + 2 + 0  =  84. 

Fichier : MOG8-03.xlsx Feuilles : S1000 et FigA 

 

 

 

 

 

 

 

 

 

 

 

 

 
 

 

Cellule Formule Copiée dans :  

B3 =LOI.NORMALE.INVERSE.N(ALEA();MoyenneX;EcartTX) B4:B1002 

C3 =SI(B3>=Garantie;"";LOI.NORMALE.INVERSE.N(ALEA();MoyenneX;EcartTX)) C3:G1002 

C1003 =1000-NB.VIDE(C3:C1002) D1003:G1003 

B1003 =SOMME(C1003:G1003) ---------------- 

 



Solutions des problèmes  7 

Les feuilles RevN et Qa du fichier MOG8-03.xlsx analysent les conséquences monétaires des 

remplacements que Celistron devra effectuer.  Dans le cas de l'échantillon de la figure ci-dessus, le 

revenu net par appareil vendu s'élève à 499,20 $ : 

 coût de la garantie  =  1678 $/u × 84 u  =  140 952 $ 

 coût de la garantie par vente  =  140 952 $  1000 u  =  140,95 $/u 

 revenu net par vente  =  2345 $/u – 1678 $/u – 140,95 $/u  =  526,05 $/u. 

 

(b) Dans un premier temps, on construit des feuilles S2000 et S5000 identiques à S1000, sauf 

que 2000 ou 5000 cas sont considérés au lieu de seulement 1000.  Puis, on calcule le nombre 

observé de télescopes de remplacement par appareil vendu (voir les cellules C4, C5 et C6 de la 

figure qui suit).  On répète l'opération 20 fois à l'aide de la commande Table de données… d'Excel. 

Les résultats obtenus sont reproduits dans le figure ci-dessous.  On observe que les quotients 

RemplT / Ventes oscillent autour de 10%.  De plus, la variabilité décroît lorsque le nombre N de 

ventes augmente : l'écart type des quotients (voir plage B26:D26) diminue de 103 à 79 et à 45 

dix-millièmes lorsque N passe de 1000 à 2000, puis à 5000. 

 

Fichier : MOG8-03.xlsx Feuilles : Qb et FigB 

 

 

 

 

 

 

 

 

 

 

 

 

 

 
 

 

 

Cellule Formule Copiée dans :  

B4 ='S1000'!B1003/1000 --------------- 

C4 ='S2000'!B2003/2000 --------------- 

D4 ='S5000'!B5003/5000 --------------- 

B25 =MOYENNE(B5:B24) C25:D25 

B26 =ECARTYPE(B5:B24) C26:D26 

 

Note.  Nous nous intéressons ici à la valeur théorique du nombre attendu de télescopes de remplacement 

par appareil vendu. Calculons d'abord la probabilité p pour qu'un télescope doive être remplacé.  

Rappelons que, par hypothèse, la durée de vie X d'un appareil de Celistron obéit à une loi normale 

N(4; 1,52).  Par conséquent, 

p  =  P(X < 2)  = P( 

𝑋−𝜇

𝜎
<

2−4

1,5
 )  = P(Z < –1,333). 

Selon les tables de la loi N(0; 1), p = 0,0913.  On peut aussi utiliser la formule suivante d'Excel, 

=LOI.NORMALE.N(2;MoyenneX;EcartTX;1) 

qui renvoie la valeur p = 0,0912112197258679. 

 

Supposons maintenant que Celistron vende N télescopes.  Une proportion p = 0,0912 d'entre eux, soit pN 

appareils, auront une durée de vie inférieure à la garantie.  Les pN télescopes de remplacement obéiront à 

la même loi de probabilité que X et p × (pN) = p2N d'entre eux tomberont en panne prématurément, ce qui 

forcera Celistron à fournir gratuitement p2N nouveaux appareils.  Ces derniers auront eux aussi une 

probabilité p de ne pas durer suffisamment longtemps et de devoir être remplacés…  En tout, le nombre 

attendu total t de télescopes requis pour honorer la garantie s'élève à  

t  =  pN + p2N + p3N + …  =  
𝑝

1−𝑝
 N  =  

0,0912

1−0,0912
 N  =  0,1004 N.   

Le coefficient 0,1004 représente le nombre attendu de télescopes de remplacement par appareil vendu.  Ce 

paramètre est estimé par le quotient RemplT / Ventes.  Dans la figure A, le quotient 84 / 1000 = 0,0084 

diffère passablement de la valeur théorique 0,1004.  Il s'agit cependant d'un cas exceptionnel et la majorité 

des estimés reproduits à la figure précédente sont beaucoup plus près de 0,1004 : par exemple, ceux de la 

colonne B varient de 0,0860 à 0,1240, pour une valeur moyenne 0,1004 qui coïncide (au niveau de 

précision considéré) avec la valeur théorique. 

 

 

 

4. La somme de deux dés 
 

La figure au haut de la page suivante indique comment nous avons procédé.  Tout d'abord, il faut 

simuler le lancer de deux dés : la formule  

1+6*ALEA() 

renvoie un nombre pseudo-aléatoire entre 1 et 7, 7 étant exclus; comme la fonction ENT efface la 

partie fractionnaire, chacun des deux termes de la formule en B3 donne un nombre entier de 1 

à 6, les 6 valeurs étant équiprobables.  En résumé, la formule en B3 représente la somme des 

points sur deux dés. 

 

Pour simuler une partie, il suffit de générer les résultats du joueur (colonne B) et du croupier 

(colonne C), puis de comparer les valeurs obtenues.  Dans la formule de la colonne D, 1 

correspond à un gain du joueur, et 0 à une perte. 

 

Dans la figure de la page suivante, nous avons simulé 2000 parties.  La formule 

=SOMME(D3:D2002) 

renvoie le nombre de gains; pour obtenir l'estimé demandé, il suffit maintenant de diviser par 

2000.  Dans l'échantillon considéré, le joueur gagne 888 fois et la proportion observée de gains 

est égale à 44,4%. 

Fichier : MOG8-04.xlsx Feuilles : Simul et Fig 

 

 

 

 

 

 

 

 

 

 

 

 
 

 

Cellule Formule Copiée dans :  

B3 =ENT(1+6*ALEA())+ENT(1+6*ALEA()) B3:C2002 

D3 =SI(C3>B3;1;0) D4:D2002 

 

Note. Nous indiquons ici comment calculer la probabilité p théorique de gain.  Notons TJ (resp. TC) la 

somme des points des deux dés lancés par le joueur (resp. par le croupier). À cause de la symétrie, les 

probabilités P(TJ < TC) et P(TJ > TC) sont égales.  Il reste à évaluer P(TJ = TC) = Eff(TJ = TC) / 362.  Le 

numérateur est la somme des effectifs des événements En = (TJ = TC = n), où n varie de 2 à 12.  Or, 

encore en vertu de la symétrie, les événements E7+i et E7–i contiennent le même nombre de résultats. Enfin, 

Eff(TJ = TC) = 2(12 + 22 + 32 + 42 + 52) + 62 = 146. 

Ainsi, Eff(TJ = TC) = ½ (362 – 146)  et  p = (362 – 146) / (2 × 362) = 0,4437. 
 

 

 

 

5. Craps. 

 

La figure 1 (voir page suivante) indique comment nous avons procédé.  La formule de B2 est 

identique à celle qui se trouve dans la cellule B3 de la figure du problème précédent; elle donne 

la somme des points lors du lancer de deux dés. 

 

À la suite du 1
er

 lancer, trois résultats peuvent survenir :  

 le joueur gagne (1 sera affiché en C2) lorsque le résultat en B2 est égal à 7 ou 11; 

 le joueur perd (0 sera affiché en C2) lorsque le résultat en B2 est égal à 2, 3 ou 12; 

 sinon, la partie continue et C sera affiché en C2. 

La formule en C2 vérifie d'abord si le joueur a gagné.  Dans le cas positif, 1 est affiché en C2 et la 

partie est terminée. Sinon, on teste s'il a perdu.  Si oui, 0 est affiché en C2 et la partie est 

terminée. Sinon, la partie se continue et C est affiché en C2. 

Fichier : MOG8-05.xlsx Feuilles :  PartieT et Fig1 

 
Cellule Formule Copiée dans :  

B2 =ENT(6*ALEA()+1)+ENT(6*ALEA()+1) -------------- 

C2 =SI(OU(B2=7;B2=11);1;SI(OU(B2=2;B2=3;B2=11);0;"C")) -------------- 

B3 =SI(OU(C2=1;C2=0;C2="");"";ENT(6*ALEA()+1)+ENT(6*ALEA()+1)) B4:B71 

C3 =SI(OU(C2=1;C2=0;C2="");"";SI(B3=B$2;1;SI(B3=7;0;"C"))) C4:C71 

 

 

Lors des lancers subséquents, il faut d'abord vérifier si la partie est encore en cours.  C'est 

pourquoi, à partir de la ligne 3, les formules des colonnes B et C débutent par un test sur la valeur 

reportée dans la cellule de la colonne C de la ligne précédente.  Considérons, à titre d'exemple, la 

formule en B5, qui s'écrit : 

=SI(OU(C4=1;C4=0;C4="");"";ENT(6*ALEA()+1)+ENT(6*ALEA()+1)) 

Si C4 contient la valeur 1 ou 0, la partie s'est terminé lors du dernier lancer et on laissera les 

cellules B5 et C5 vides, pour indiquer que le joueur n'a plus à lancer les dés.  De même, si C4 est 

vide, la partie est terminé depuis quelque temps et, dans ce cas également, on laissera les cellules 

B5 et C5 vides. Sinon, la partie est encore en cours et le joueur doit lancer les dés une fois de 

plus.  La cellule B5 contiendra la somme des points sur les deux dés; et C5, la conclusion de ce 

lancer, soit 1 (gagne quand B5 coïncide avec le résultat en B2), 0 (perd si le joueur a obtenu un 

total de 7) ou C (sinon, la partie se continue).   

 

Nous avons réservé un maximum de 70 lignes pour noter les résultats des différents lancers.   La 

probabilité qu'il faille plus de 70 lancers dans une partie donnée est extrêmement faible.  

D'ailleurs, nous n'avons observé aucune partie qui exigeait plus de 40 lancers; le maximum de 

lancers pour une simulation de 10 000 parties se situe le plus souvent entre 25 et 30. 

 

La figure 2 (voir page suivante) résume les résultats de 10 000 parties.  La colonne B indique si le 

joueur a gagné ou perdu; la colonne C donne le nombre de fois qu'il a lancé les dés.  

Fichier : MOG8-05.xlsx  Feuilles : Simul et Fig2 

 

 

 

 

 

 

 

 

 

 

 

 
 

 

Cellule Formule 

B3 =MAX(PartieT!C2:C71) 

C3 =NB.SI(PartieT!C2:C71;"C")+1 

 

Dans l'échantillon de 10 000 parties reproduit à la figure ci-dessus, le joueur a gagné 4954 fois.  

Les feuilles Résultats et Fig3 du chiffrier MOG8-05.xlsx décrivent la distribution du nombre L de 

lancers pour les 10 000 parties d'une simulation.  Dans le cas traité à la figure 2, il faut en 

moyenne 3,49 lancers; et 95% des parties en exigent 9 ou moins. 

 

 

6. La grippe aviaire. 

 

La première étape consiste à simuler une commande de 25 millions de doses.  La figure ci-

dessous (voir page suivante) illustre une façon de procéder.  Les paramètres utilisés sont : 

 ValeursD et CumulD sont des plages qui contiennent, la première, la liste des valeurs de la 

variable aléatoire D, «durée (en jours) de fabrication d'un lot», et la seconde, la suite des 

probabilités que D soit inférieure aux différentes valeurs de D; par exemple,  

(3
e
 élément de CumulD)  =  P(D < 8)  =  P(D = 6) + P(D = 7)  =  0,04 + 0,08 = 0,12. 

 Les noms NMin et NMax renvoient à des cellules de la feuille Données qui contiennent les valeurs 

minimale 800 000 et maximale 1 500 000 de la variable aléatoire N, «nombre de doses fournies 

par un lot»; Nk = (NMod – NMin) / (NMax – NMin) représente la proportion de l'intervalle 

[NMin; NMax] située à la gauche de la valeur modale NMod = 1 000 000. 

 ProbR = 0,1 est la probabilité qu'un lot soit rejeté lors de l'inspection. 

 TComm = 25 000 000 est la taille de la commande. 

 

Chaque ligne de la figure 1 correspond à un lot.  Analysons plus en détail les données du premier 

(ligne 3).  La formule en B3 est analogue à celle apparaissant dans la cellule D4 de la figure 8.5 

du manuel (voir page 418) : elle commande à Excel de générer un nombre pseudo-aléatoire entre 

0 et 1, puis de le transformer en une valeur de la variable D.  Le résultat représente la durée 

aléatoire de fabrication du lot.  De même, C3 et D3 génèrent une valeur pseudo-aléatoire de la 

variable N qui, rappelons-le, obéit à une loi triangulaire; la formule en D3 est une adaptation au 

présent contexte de celle utilisée en section 8.2.4 (voir figure 8.19, colonne G).  Enfin, F3 

détermine, avec une probabilité ProbR = 10%, si le lot sera rejeté lors de l'inspection : dans le cas 

positif, on reportera en G3 le nombre de doses calculé en D3; sinon, le stock cumulé sera nul.  La 

dernière colonne gère une horloge, qui indique combien de jours se sont écoulés depuis le début 

des travaux du laboratoire sur la commande. 

 

Fichier : MOG8-06.xlsx Feuilles : CommandeT et Fig1 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 
 

 

Cellule Formule 

B3 =RECHERCHE(ALEA();CumulD;ValeursD) 

C3 =ALEA() 

D3 =ENT(SI(C3<=Nk;NMin+(NMax-NMin)*RACINE(Nk*C3);NMax-(NMax-NMin)*RACINE((1-Nk)*(1-C3)))) 

E3 =SI(ALEA()<ProbR;0;1) 

F3 =SI(E3=1;D3;0) 

G3 =B3 

B4 =SI(OU(F3>=TComm;F3=" ");" ";RECHERCHE(ALEA();CumulD;ValeursD)) 

G4 =SI(OU(F3>=TComm;F3=" ");" ";G3+B4) 

 

Les formules de la ligne 3 sont copiées – avec un ajustement dont nous parlons ci-dessous – dans 

les lignes subséquentes.  Nous avons prévu un maximum de 50 lignes ou lots pour une 

commande de 25 millions de doses.  Il est fort improbable que ce maximum soit excédé et le 

risque que sa présence dans notre modèle nous induise en erreur est minime.  L'ajustement 

mentionné ci-dessus tient au fait qu'il faut tenir compte de la possibilité que la commande soit 

déjà complétée; dans ce cas, on voudra ne rien afficher dans la ligne correspondante.  

Considérons, à titre d'exemple, la formule qui a été reportée en B7 : 

=SI(OU(F6>=TComm;F6=" ");" ";RECHERCHE(ALEA();CumulD;ValeursD)). 

Si les 25 millions de doses requises ont été obtenues avec le lot précédent (F6>=TComm), il est 

inutile de fabriquer un lot additionnel et les travaux du laboratoire s'arrêtent : nous convenons de 

laisser vides les colonnes B à G de la ligne courante (ligne 7).  De même, si la condition 

« F6="" » est satisfaite, c'est que la commande a été complétée antérieurement et, de nouveau, nous 

voulons ne rien afficher sur la ligne courante.  Enfin, dans tous les autres cas, les travaux du labo-

ratoire doivent se poursuivre et Excel renverra en B7 le temps de fabrication d'un nouveau lot. 

 

La 2
e
 étape vise à simuler 1000 commandes.  Concrètement, nous avons reporté la formule 

=MAX(CommandeT!G3:G52) 

dans la cellule B2 de la feuille Simul; puis, nous avons sélectionné la plage A2:B1002 et avons 

utilisé la commande Table de données… d'Excel. 

 

La figure 2 résume les résultats obtenus.  Selon notre échantillon, la Ministre devra tenir compte 

du fait que le temps requis pour une commande s'échelonne de 196 à 352 jours.  De plus, le délai 

moyen est de 257,5 jours, avec un écart type de 21,5 jours. 

Fichier : MOG8-06.xlsx      Feuilles : Résultats et Fig2 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

7. Les tapis roulants. 

 

La figure 1 (voir page suivante) indique comment nous avons procédé.  Les formules de la plage 

B3:G3 sont copiées dans B4:G1002.  Les paramètres utilisés sont : 

 Les noms DMin et DMax renvoient à des cellules de la feuille Données qui contiennent les valeurs 

minimale 2 000 et maximale 6 000 de la variable aléatoire D, «demande (en u) durant la période 

ciblée de deux mois»; Dk = (DMod – DMin) / (DMax – DMin) représente la proportion de 

l'intervalle [DMin; DMax] située à la gauche de la valeur modale DMod = 3 500. 

 PrMin = 500 et PrMax = 700 sont les valeurs minimale et maximale de la variable aléatoire Pr, 

«prix de vente (en $) durant la période».  

 QtéF = 5 000 est le nombre d'unités qui seront fabriquées. 

 CoûtU = 345 est le coût de revient (en $/u). 

Fichier : MOG8-07.xlsx Feuilles :  Simul et Fig1 

 
Cellule Formule 

B3 =ALEA() 

C3 =ARRONDI(SI(B3<=Dk;DMin+(DMax-DMin)*RACINE(Dk*B3);DMax-(DMax-DMin)*RACINE((1-Dk)*(1-B3)));0) 

D3 =MIN(C3;QtéF) 

E3 =ARRONDI(PrMin+ALEA()*(PrMax-PrMin);2) 

F3 =QtéF-D3 

G3 =(D3*E3)+(F3*ARRONDI(E3/2;2))-QtéF*CoûtU 
 

Les formules de B3 et C3 génèrent une valeur pseudo-aléatoire de la variable D qui, rappelons-le, 

obéit à une loi triangulaire; la formule en C3 est une adaptation au présent contexte de celle 

utilisée en section 8.2.4 (voir figure 8.19, colonne G); enfin, le recours à la fonction ARRONDI 

d'Excel découle du fait que la demande D est nécessairement un nombre entier.  Les ventes ne 

peuvent évidemment dépasser ni la demande, ni la quantité fabriquée, ce qui justifie de définir en 

colonne D les ventes comme le minimum de la valeur en colonne C et de QtéF.  La formule de la 

colonne E génère une valeur pseudo-aléatoire de la variable Pr qui obéit à une loi uniforme; le 

prix obtenu est arrondi à 2 décimales, c'est-à-dire à la cent près.  La colonne F contient le nombre 

d'unités soldées, qui tout naturellement se définit comme la différence entre le nombre d'unités 

fabriquées et le nombre d'unités vendues.  Enfin, le profit qu'apportera à notre fabricant la vente 

du lot de 5 000 tapis est calculé en colonne G comme la différence entre les revenus et les coûts :  

 Le coût total encouru par le fabricant s'obtient comme le produit du nombre d'unités 

fabriquées, QtéF = 5 000, et du coût unitaire de revient, CoûtU. 

 Le revenu provenant des tapis vendus au prix régulier est égal au produit de la demande 

(colonne D) et du prix unitaire (colonne E). 

 On procède de façon analogue pour les tapis soldés à moitié prix : le nombre d'unités se 

trouve cette fois en colonne F, tandis que le prix unitaire se calcule en divisant par 2 le prix 

régulier (colonne E), puis en arrondissant à la cent près. 

La figure 2 (voir page suivante) résume les résultats obtenus.  Selon notre échantillon, le revenu 

pourrait s'établir entre un minimum de 125 799 $ et un maximum de 1 769 300 $; par ailleurs, il y 

a 50% des chances pour que le lot de 50 000 tapis rapporte plus de 909 517 $ au fabricant, et 50% 

des chances pour qu'il en rapporte moins.  Enfin, le revenu espéré s'élève à 922 793 $. 

Fichier : MOG8-07.xlsx Feuilles : Résultats et Fig2 

 
 

 

 

 

8. Un appel d’offres. 

 

La figure 1 ci-dessous indique comment nous avons procédé.  Les paramètres utilisés sont : 

 CoûtP = 12 100 est le coût estimé du projet (en k$). 

 Les noms CMin et CMax renvoient à des cellules de la feuille Données qui contiennent les valeurs 

minimale 0,9 × CoûtP = 10 890 et maximale 2,5 × CoûtP = 30 250 de la variable aléatoire C, 

«montant (en k$) de la soumission»; Ck = (CMod – CMin) / (CMax – CMin) représente la 

proportion de l'intervalle [CMin; CMax] située à la gauche de la valeur modale CMod = 

1,3 × CoûtP = 15 730. 

 SoumP représente le montant (en k$)de la soumission de Panzetta; ce paramètre, qui se 

trouve dans la cellule A3 de la feuille Résultats (voir figure 2), sert à relier le tableau de cette 

feuille aux calculs de la feuille Simul. 

 

Fichier : MOG8-08.xlsx Feuilles :  Simul et Fig1 

 
Cellule Formule Copiée dans :  

B3 =ALEA() B4:B5002 

J3 =MIN(C3;E3;G3;I3) J4:J5002 

K3 =SI(SoumP<=J3;1;0) K4:K5002 

Les colonnes B à I génèrent des valeurs pseudo-aléatoires pour les soumissions des quatre 

concurrents de Panzetta.  En pratique, nous avons d'abord reporté en B3 et en C3 les formules 

pertinentes, puis nous avons copié la plage B3:C3 dans B3:I5002.  La cellule C3 transforme le 

nombre pseudo-aléatoire de B3 en une valeur de la variable C, qui, rappelons-le, obéit à une loi 

triangulaire.  La formule de C3, qui se lit : 

=SI(B3<=Ck;CMin+(CMax-CMin)*RACINE(Ck*B3);CMax-(CMax-CMin)*RACINE((1-Ck)*(1-B3))) 

est une adaptation au présent contexte de celle utilisée en section 8.2.4 du manuel.   Noter que, 

par exemple, la formule de E3 réfère à la cellule D3, et non à B3; en effet, les concurrents 

doivent préparer leur soumission de façon indépendante, sinon ils pourraient être accusés de 

collusion. 

 

Pour déterminer si Panzetta obtient le contrat, il reste maintenant à vérifier si sa soumission est la 

plus basse.  C'est l'objet des colonnes J et K : il suffit de comparer le montant SoumP exigé par 

Panzetta dans son offre au maximum des montants apparaissant dans les colonnes C, E, G et I 

(les cas d'égalité, qui devraient être rarissimes, sont tranchés ici en faveur de Panzetta).  La 

formule de la colonne K renvoie 1 quand Panzetta obtient le contrat, et 0 sinon. 

 

La comparaison des différentes options qui s'offrent à Panzetta se fait dans les feuilles Résultats et 

Fig2, dont des extraits sont reproduits dans la figure 2 (voir page suivante).   

 

Tout d'abord, nous avons reporté, dans la plage A3:C3 de la feuille Résultats, la valeur du 

paramètre SoumP utilisé dans la feuille Simul (A3), la proportion des 5000 cas de la simulation 

où Panzetta décroche le contrat (B3) et le revenu net moyen de Panzetta pour ces 5000 cas 

(C3). Puis, nous avons sélectionné la plage A3:C16 et utilisé la commande Table de données… 

d'Excel. 

 

La meilleure stratégie est évidemment celle qui correspond au revenu le plus élevé.  Pour mettre 

la réponse en évidence, nous avons calculé en C17 la valeur maximale de la plage C4:C16, puis 

nous avons comparé le contenu de C17 avec les différentes valeurs apparaissant dans cette plage.  

Selon l'échantillon considéré aux figures 1 et 2, Panzetta aurait intérêt à fixer sa soumission à 

14 750 000 dollars. 

Fichier : MOG8-08.xlsx Feuilles : Résultats et Fig2 

 
Cellule Formule Copiée dans :  

B3 =MOYENNE(Simul!K3:K5002) -------------- 

C3 =(A3-CoûtP)*B3 -------------- 

C17 =MAX(C4:C16) -------------- 

D4 =SI(C4=C$17;A4;"") D5:D16 

 

 

 

 

9. Les soles du restaurant Les Deux Gauloises 

 

Dans un premier temps, nous simulons 1000 journées.  Les paramètres utilisés sont : 

 ValeursD et CumulD sont des plages qui contiennent, la première, la liste des valeurs de la 

variable aléatoire D, «demande», et la seconde, la suite des probabilités que D soit inférieure 

aux différentes valeurs de D; par exemple,  

(3
e
 élément de CumulD)  =  P(D < 60)  =  P(D = 40) + P(D = 50)  =  0,10 + 0,15 = 0,25. 

 CtP, qui représente le coût de pénurie (en $), sera fixé à 0 dans la question (a), mais variera 

de 0,50 $ à 2 $ par pas de 0,25 $ en question (b). 

 

La figure ci-dessous (voir page suivante), qui reproduit des extraits des feuilles Simul et FigS, 

illustre comment calculer les revenus nets de 1000 jours quand le coût de pénurie CtP est fixé à 

2 $.  Noter que, pour obtenir les formules des colonnes D à M, il suffit de saisir celles des cellules 

D3 et E3, puis de copier la plage D3:E3 dans D3:M1002. 

Fichier : MOG8-09.xlsx Feuilles : Simul et FigS 

 
Cellule Formule Copiée dans :  

B3 =ALEA() B4:B1002 

C3 =RECHERCHE(B3;CumulD;ValeursD) C4:C1002 

D3 =MIN($C3;E$1) colonnes D, F, H, J et L 

E3 =20*D3+6*(E$1-D3)-10*E$1-CtPén*($C3-D3) colonnes E, G, I, K et M 

 

 

La colonne C transforme les nombres pseudo-aléatoires de la colonne B en valeurs de la variable 

D, tandis que les colonnes D à M donnent les ventes et le revenu net d'une journée selon les cinq 

scénarios envisagés par le restaurateur.  Évidemment, le chiffre des ventes (colonne D) est limité 

à la fois par le demande (colonne C) et par le nombre de soles commandées (cellules E1, G1, I1, 

K1 et M1); il est donc raisonnable de reporter en D3 la formule =MIN($C3;E$1).  Quant au revenu 

net, il se calcule comme la différence entre les revenus et les coûts.  Considérons de plus près la 

1
re

 journée de l'échantillon de la figure (ligne 3) et commentons brièvement les calculs des 

cellules E3 et M3.  Le demande (C3) sera de 70 soles.  Selon le 1
er

 scénario, les achats (E1) sont 

de 40 soles et le revenu net (E3) s'élèvera à 340 $ : 

 Le propriétaire aura déboursé (10 × 40) = 400 dollars pour l'achat des 40 poissons. 

 Entre le jeudi et le samedi, le restaurant vendra 40 soles, pour un revenu de (20 × 40) = 800 

dollars. 

 30 des 70 clients commandant une sole n'obtiendront pas le plat choisi et le restaurateur 

encourra un coût de pénurie de (2 × 30) = 60 dollars. 

 En résumé, RevN = 800 – 400 – 60 =  340. 

 

De même, selon le 5
e
 scénario, les achats (M1) sont de 80 soles et le revenu net (M3) s'élèvera 

à 660 $ : 

 Le propriétaire aura déboursé (10 × 80) = 800 dollars pour l'achat des 80 poissons. 

 Entre le jeudi et le samedi, le restaurant vendra 70 soles, pour un revenu de (20 × 70) = 1400 

dollars. 

 Les 10 poissons en excédent entreront dans les plats cuisinés et rapporteront (6 × 10) = 60 

dollars. 

 En résumé, RevN = 1400 – 800 + 60 =  660. 

(a) Le tableau suivant donne le revenu net moyen pour un échantillon de 1000 journées en 

fonction du nombre TComm de soles commandées. 

 

TComm 40 50 60 70 80 

RevN moyen 400,0 488,0 551,0 581,0 566,6 

 

On constate que, pour cet échantillon, la meilleure stratégie pour le restaurateur est de fixer à 70 

soles sa commande hebdomadaire. 

 
Note.  Le revenu net, quand TComm = 40, se calcule comme suit : 

RevN  =  40 × (20 – 10) – (CtPén × D)  =  400 – (CtPén × D), 

où D est la demande.  Si CtPén = 0, le revenu net est une constante. 

 

 

(b) La figure ci-dessous indique comment nous avons procédé pour analyser l'impact du coût 

de pénurie sur le revenu moyen.  Nous avons d'abord reporté la formule 

=MOYENNE(Simul!E3:E1002) 

dans la cellule B8 et l'avons copiée dans C8:F8; puis, nous avons complété les plages A9:A17 et 

B7:F7.  Ensuite, nous avons sélectionné la plage A2:B1002 et utilisé la commande Table de 

données… d'Excel.  Enfin, les colonnes G et H détermine la stratégie optimale, compte tenu du 

coût de pénurie retenu par le restaurateur. 

 

Fichier : MOG8-09.xlsx Feuilles :  Qb et FigB 

 
Cellule Formule Copiée dans :  

B8 =MOYENNE(Simul!E3:E1002) C8:F8 

G9 =MAX(B9:F9) G10:G17 

H9 =SI(G9=B9;B$7;SI(G9=C9;C$7;SI(G9=D9;D$7;SI(G9=E9;E$7;F$7)))) H10:H17 

 

 

La meilleure stratégie, selon notre échantillon, consiste à commander 70 soles par semaine, 

quelle que soit la valeur du coût de pénurie. 

10. Le marketing des petits pois. 

 

La première étape de notre analyse consiste à simuler une période typique de 40 trimestres 

consécutifs.  La figure 1 ci-dessous décrit une façon de procéder dans le cas où le producteur P2 a 

mené une campagne normale durant le dernier trimestre. 

 

Fichier : MOG8-10.xlsx Feuilles :  PériodeT et Fig1 

 
Cellule Formule Copiée dans :  

B4 =SI(C3=1;SI(ALEA()<=0,7;1;0);SI(ALEA()<=0,6;1;0)) B5:B43 

C4 =SI(C3=1;SI(ALEA()<=0,4;1;0);SI(ALEA()<=0,5;1;0)) C5:C43 

D4 =INDEX(Tr;2*$B4+$C4+1;1) D5:D43 

E4 =INDEX(Tr;2*$B4+$C4+1;2) E5:E43 

F4 =INDEX(Tr;2*$B4+$C4+1;3) F5:F43 

G4 =INDEX(Tr;2*$B4+$C4+1;4) G5:G43 

H4 =ALEA() H5:H43 

I4 =SI(H4<=G4;D4+(F4-D4)*RACINE(G4*H4);F4-(F4-D4)*RACINE((1-G4)*(1-H4))) I5:I43 

J4 =MAX(MIN(J3+I4;1);0) J5:J43 

 

Dans cette figure, on a associé une ligne à chacun des trimestres, la ligne 3 (t = 0) correspondant 

au trimestre précédant immédiatement la période considérée.  
 

 Les colonnes B et C donnent la stratégie publicitaire des deux concurrents pour un trimestre 

donné, 1 représentant une campagne intense et 0, une campagne normale.  Ainsi, la valeur 0 

reportée en C3 indique que, dans la feuille Fig1, nous prenons comme hypothèse que P2 a 

mené une campagne publicitaire normale lors du trimestre t = 0.  Analysons plus en détail la 

formule de B4: 

=SI(C3=1;SI(ALEA()<=0,7;1;0);SI(ALEA()<=0,6;1;0)) 

D'après l'énoncé du problème, P1 choisit une campagne intense avec une probabilité de 70% 

quand la campagne de P2 fut intense le trimestre précédent, et avec une probabilité de 60% 

dans le cas contraire.  Conformément à cette règle, en B4, on teste d'abord la décision de P2 

lors du trimestre 0; dans le cas d'une campagne intense (valeur 1 en B3), Excel renverra la 

valeur 1 avec une probabilité de 70%; sinon, la valeur 1 sera obtenue avec une probabilité 

de 60%.  De même, la colonne C traduit la stratégie de P2 décrite dans l'énoncé. 
 

 Par hypothèse, la variable Xt, « changement de la part de marché de P1 lors du trimestre t », 

obéit à une loi triangulaire dont les paramètres dépendent du type de campagne publicitaire 

menée par les deux concurrents.  Des valeurs pseudo-aléatoires de Xt sont obtenues de la 

façon suivante : on génère d'abord, en colonne H, un nombre pseudo-aléatoire entre 0 et 1; 

puis, en colonne I, ce dernier est transformé en une valeur d'une loi triangulaire à l'aide d'une 

version adaptée au présent contexte de la formule apparaissant dans la cellule G6 de la figure 

8.19 (voir page 441). Afin d'alléger, nous avons reporté dans les colonne D à G les 

paramètres pertinents de la loi triangulaire.  Ainsi, dans les formules des cellules D4 à G4 de 

la figure 1, le nom Tr renvoie à un tableau 4 × 4 de la feuille Données, dont les lignes 

correspondent aux 4 combinaisons possibles de campagne publicitaire par les deux 

concurrents; les trois premières colonnes de Tr contiennent les valeurs minimales, modales et 

maximales indiquées dans le 1
er

 tableau de l'énoncé, et la dernière, les valeurs du paramètre k 

(voir p. 441, ligne 1; ou encore, voir la cellule B3 de la figure 1 de la solution du 

problème 6).  La fonction Index(Tr;i;j) d'Excel renvoie l'entrée du tableau Tr située à 

l'intersection de la ligne i et de la colonne j : en D4 est donc affichée la valeur minimale 

(j = 1) de la ligne numéro i = (2 × 0) + 0 + 1 = 1, « Ni P1, ni P2 »; en E4 et F4, on retrouve les 

valeurs modale et maximale de cette même ligne; enfin, la plage D43:G43 contient les quatre 

paramètres de la ligne numéro i = (2 × 1) + 0 + 1 = 3, « Seulement P1 ». 
 

 Pour obtenir la part de marché Xt de P1 à la fin du trimestre t, il suffit d'ajouter le 

changement survenu durant le trimestre à la part de marché à la fin du trimestre précédent.  

La partie centrale de la formule de la colonne J traduit cette opération.  Cependant, comme 

une part de marché ne peut ni être négative ni excéder 1, on force la valeur de Xt à rester 

entre les limites 0 et 1 à l'aide des fonctions Max et Min d'Excel. 

 

L'énoncé demande de simuler 1000 périodes de 40 trimestres et de compiler les valeurs de Xt où  

t = 4, 8 et 40.  Il suffit de compléter la plage B3:D3 comme dans la figure 2 ci-dessous, puis de 

recourir à la commande Table de données… d'Excel. 

 

Fichier : MOG8-10.xlsx Feuilles : Simul et Fig2 

 
Cellule Formule 

B3 =PériodeT!J7 

C3 =PériodeT!J11 

D3 =PériodeT!J43 

 

La plage B3:D11 de la feuille Fig3 (voir page suivante) résume les statistiques des variables X4, 

X8 et X40 dans l'échantillon considéré dans la figure précédente.  Pour obtenir les résultats des 

colonnes E à G, nous avons reporté la valeur 1 dans le cellule de la feuille PériodeT, puis avons 

recalculé le tableau de la feuille Simul en pressant la touche F9. 

Fichier : MOG8-10.xlsx Feuilles : Résultats et Fig3 

 
 

Les distributions des variables X4, X8 et X40 ne diffèrent pas significativement selon que la 

campagne menée par P2 lors du trimestre 0 fut normale ou intense.  La part moyenne de marché 

de P1 tend à augmenter avec le temps : elle est d'environ 57,7% à la fin du trimestre 4 (voir 

cellules B8 et E8), mais atteint 59,5% un an plus tard et 73,3% à la fin de la période considérée.  

Cependant, la variabilité augmente également avec le temps et, à long terme, la part de marché de 

P1 pourrait, à la limite, tomber aussi bas que 22,4%.  Évidemment, les résultats à long terme 

reposent sur les hypothèses qu'aucun changement structurel ne survienne dans le marché des 

petits pois et que les deux concurrents conservent leur politique de marketing pendant les 40 

prochains trimestres. 

 

 

11. Le quotidien Les Échos. 

 

La figure 1 ci-dessous (voir page suivante) indique comment nous avons procédé.  Les 

paramètres utilisés sont : 
 

 PrixV = 1,00 et CtAch = 0,45 sont le prix de vente et le coût d'achat d'un exemplaire. 

 CtPén = 0,20 représente le coût de pénurie. 

 

On génère, en colonne B, un nombre pseudo-aléatoire entre 0 et 1; puis, en colonne C, ce dernier 

est transformé en une valeur d'une loi triangulaire à l'aide d'une version adaptée au présent 

contexte de la formule apparaissant dans la cellule G6 de la figure 8.19 (voir page 441).  Plus 

précisément, on a reporté en C4 la formule suivante, qu'on a ensuite copiée dans la plage 

C5:C2003 : 

=ARRONDI(SI(B4<=Xk;XMin+(XMax-XMin)*RACINE(Xk*B4);XMax-(XMax-XMin)*RACINE((1-Xk)*(1-B4)));0) 

Ici, les noms XMin et XMax renvoient à des cellules de la feuille Données qui contiennent les valeurs 

minimale 140 et maximale 250  de la variable aléatoire X, «nombre d'exemplaires vendus un jour 

donné»; Xk = (XMod – XMin) / (XMax – XMin) représente la proportion de l'intervalle [XMin; XMax] 

située à la gauche de la valeur modale XMod = 225.  

Les colonnes D à J donnent le revenu de la journée en fonction du nombre d'exemplaires achetés 

par le propriétaire du dépanneur.  Nous expliquons brièvement la formule reporté en D4 : 

 

 Le revenu de la journée est le produit du prix de vente unitaire, PrixV, et du nombre V 

d'exemplaires vendus; et V s'obtient comme le minimum de la demande (C4) et du nombre 

d'exemplaires disponibles (D3). 
 

 Le propriétaire du dépanneur doit débourser 0,45 $ par exemplaire acheté; au total, il versera 

0,45 × 205 = 92,25 dollars.  De façon à obtenir une formule qui s'adapte correctement lors 

de la copie dans la plage D4:J2003, on récrit ce produit sous la forme équivalente « CtAch*D$3 ». 
 

 Le coût de pénurie total associé à une journée se calcule comme le produit du coût unitaire 

CtPén = 0,20 et du nombre M d'exemplaires manquants.  Par définition, M est égal à l'écart 

entre la demande des clients (C4) et le nombre d'exemplaires achetés par le dépanneur (D3), 

en autant que la demande excède l'offre; sinon, M est nul. 

 

Fichier : MOG8-11.xlsx Feuilles : Simul et Fig1 

 
Cellule Formule Copiée dans :  

B4 =ALEA() B5:B2003 

D4 =PrixV*MIN(D$3;$C4)-CtAch*D$3-CtPén*MAX(0;$C4-D$3) D4:J2003 

 

 

Le revenu quotidien moyen (en $) des 2000 journées de notre échantillon est donné dans la 

figure 2 en fonction du nombre d'exemplaires achetés.  On constate que la meilleure politique 

consiste à acheter entre 215 et 220 copies par jour.  (Il est inutile de chercher à être plus précis, à 

cause de la variabilité des résultats de la simulation.) 

 

Fichier : MOG8-11.xlsx Feuilles : Résultats et Fig2 

 
Cellule Formule Copiée dans :  

B6 =MOYENNE(Simul!D4:D1003) C6:H6 

12. Le juste-à-temps. 

 

(ab) Noter que, si le quai est desservi par 3 équipes et que 3 camions surviennent en début 

d’heure alors que 2 camions sont déjà en attente, ces deux derniers ainsi qu’un seul des 3 nouveaux 

arrivés seront déchargés durant l'heure suivante et les 2 autres attendront exactement une heure. 

 

Dans un premier temps, nous simulons une période typique de 100 jours consécutifs, soit 2400 

heures d'opération en continu.  Les paramètres Valeurs et CumulX renvoient à des plages qui 

contiennent, la première, la liste des valeurs de la variable aléatoire X, «nombre de camions qui 

arrivent au quai de débarquement durant une heure donnée», et la seconde, la suite des 

probabilités que X soit inférieure aux différentes valeurs de X; par exemple,  

(3
e
 élément de CumulX)  =  P(X < 2)  =  P(X = 0) + P(X = 1)  =  0,05 + 0,10  = 0,15. 

La colonne B génère une valeur pseudo-aléatoire de la variable X ; celle-ci est ajoutée en 

colonnes C et E au nombre de camions en attente à la fin de l'heure précédente.  De ce total, on 

retranche le nombre d'équipes, 3 ou 4 (voir cellules B4 et D4), pour déterminer combien de 

camions attendront durant l'heure courante. 

 

Fichier : MOG8-12.xlsx   Feuilles : PériodeT et Fig1 

 
Cellule Formule Copiée dans :  

B4 =RECHERCHE(ALEA();CumulX;Valeurs) B5:B2403 

C4 =D3+$B4 C5:C2403 et E5:E2403 

D4 =MAX(0;C4-3) D5:D2403 

F4 =MAX(0;E4-4) F5:F2403 

 

 

La ligne 4 de la figure 2 (voir page suivante) résume les résultats obtenus à la figure 1.  En 

particulier, les cellules C4 et E4 donnent le coût moyen selon qu'il y a 3 ou 4 équipes : il suffit 

de multiplier le nombre moyen de camions en attente par la pénalité Pén = 150 encourue par la 

manufacture quand un camion doit attendre 1 heure en file qu’on le décharge.  La période 

typique est ensuite répétée 1000 fois à l'aide de la commande Table de données… d'Excel.  

Fichier : MOG8-12.xlsx Feuilles : SimulP et Fig2 

 
Cellule Formule Copiée dans :  

B4 =MAX(PériodeT!D3:D2403) D4 

C4 =Pén*MOYENNE(PériodeT!D4:D2403) E4 

 

La figure 3 donne les principales statistiques des deux variables « Coût » considérées à la figure 

précédente. Dans notre échantillon de 1000 périodes typiques, le coût encouru par la manufacture 

s'élève en moyenne à 151 $/h quand il y a 3 équipes, et à seulement 9,71 $/h quand il y en a 4.  

La direction ajoutera ou non une 4
e
 équipe en tenant compte à la fois du coût de cette équipe 

additionnelle et des inconvénients intangibles de la situation actuelle où de nombreux camions 

doivent attendre d'être déchargés, parfois aussi longtemps que 5 heures : en effet, tel qu’indiqué 

par exemple dans la cellule B9 de la figure 2, la file peut atteindre jusqu’à 13 véhicules à un 

moment donné ; les 13 attendront d’abord 1 heure, puis 12 seront déchargés par les 3 équipes 

pendant les 4 heures suivantes, tandis que le dernier devra attendre le début de la 6
e
 heure. 

 

Fichier : MOG8-12.xlsx    Feuilles : RésultP et Fig3 

 

 

 

 (c) La figure 4 (voir page suivante) décrit une journée typique dans ce nouveau contexte.  Les 

calculs sont semblables à ceux de la figure 1, sauf qu'on recourt dans les lignes 4 et 19 aux 

probabilités des variables Y et D respectivement, plutôt qu'à celles de X.  

Fichier : MOG8-12.xlsx Feuilles : JourT et Fig4 

 
Cellule Formule Copiée dans :  

B4 =RECHERCHE(ALEA();CumulY;Valeurs) --------------- 

B5 =RECHERCHE(ALEA();CumulX;Valeurs) B6:B19 

B19 =RECHERCHE(ALEA();CumulD;Valeurs) --------------- 

C4 =D3+$B4 C5:C19  et E5:E19 

D4 =MAX(0;C4-3) D5:D19 

F4 =MAX(0;E4-4) F5:F19 

On procède ensuite comme aux figures 2 et 3.  On calcule d'abord le coût total de la journée 

typique, puis on simule 2000 jours (voir figure 5 ci-dessous).  Enfin, on calcule les principales 

statistiques de la variable « Coût total » : dans notre échantillon de 2000 jours, la manufacture 

doit rembourser aux fournisseurs en moyenne 1953 $/j quand il y a 3 équipes, et seulement 

171 $/j quand il y en a 4.   

Fichier : MOG8-12.xlsx Feuilles : SimulJ et Fig5 

 
Cellule Formule Copiée dans :  

B4 =MAX(JourT!C4:C19) D4 

C4 =Pén*SOMME(JourT!D4:D19) E4 

13. Le tir aux pigeons. 

 

Dans un premier temps, nous analysons une séance de tir typique.  La figure 1 (voir page 

suivante) indique comment nous avons procédé.  Noter que nous avons masqué les colonnes C, 

D, F et I afin d’alléger la figure. De plus, les formules des colonnes J et K sont copiées de celles 

des colonnes G et H respectivement; elles ne sont pas affichées, toujours pour alléger. Les 

paramètres utilisés dans les formules sont : 

 Les noms RMin, RMod et RMax renvoient à des cellules de la feuille Données qui contiennent les 

valeurs minimale 5, modale 7 et maximale 12 de la variable R, « temps requis (en secondes) pour 

recharger et se mettre en position de tir ». 

 De même, VMin = 1, VMod = 2 et VMax = 5 représentent les valeurs minimale, modale et 

maximale de la variable V, « temps requis (en secondes) pour viser et tirer ». 

 Enfin, les valeurs minimale et maximale de la variable L, « temps (en secondes) entre deux 

lancers consécutifs de pigeons », sont LMin = 7 et LMax = 13 respectivement. 

 p = 0,8 est la probabilité d'atteindre un pigeon lors d'un tir. 

 NbP = 20 est le nombre de pigeons que le tireur veut abattre lors de la séance. 

 

Les lignes 43 à 53 de la figure 1 n'affichent que le numéro de la colonne A, car le tireur a atteint 

son objectif de 20 pigeons lors du 39
e
 lancer (voir cellule M42) et la séance s'est terminée à ce 

moment-là.  Comme le nombre de lancers requis n'est pas fixe mais dépend du succès rencontré 

par le tireur, nous devons prévoir suffisamment de lignes pour être assurés que la séance sera 

complétée à la fin du tableau.  En pratique, nous en avons réservé 50.  La probabilité qu'il faille 

plus de 50 lancers lors d'une séance est extrêmement faible; d'ailleurs, le maximum observé dans 

notre simulation de 1000 séances est de 43 lancers (voir figure 3, cellule B7).  

 

Les formules de la figure 1 débutent par la fonction Si et renvoient une cellule vide quand la 

séance est terminée.  Dans les colonnes C à M, on se contente de considérer le contenu de la 

cellule en colonne B de la même ligne : si celle-ci est vide, la séance a déjà pris fin, et notre 

conclusion sera de laisser vides également les autres cellules de la ligne.  Noter que, dans les 

formules de G4 et de H4, la référence à B4 se fait indirectement, en vérifiant le contenu de la 

cellule F4 immédiatement à gauche.  Le test de la colonne B est plus complexe et mérite quelques 

commentaires additionnels.  Par exemple, en B5, nous testons si l'objectif de 20 pigeons a été 

atteint lors du tir précédent (M4=NbP) ou lors d'un tir antérieur (M4=""); comme aucune de ces 

conditions n'est satisfaite, la séance se continue et il faut simuler un autre lancer.  Par contre, en 

B43, le test s'écrit OU(M42=NbP;M42=""); dans la simulation reproduite à la figure 1, le test donne 

une réponse positive puisque la ligne précédente contient le nombre 20 en colonne M (M42=20) ; 

par conséquent, la cellule B43 sera vide.  De même, en B44, le test consiste à évaluer 

OU(M43=NbP;M43="") et, cette fois, c’est la seconde condition qui est satisfaite puisque la cellule 

M43 est vide. Enfin, le test en I4 sera commenté un peu plus loin ci-dessous. 

 

Nous analysons maintenant les formules des colonnes B à M dans le cas où, la conclusion du test 

étant négative, il faut poursuivre la séance. 

Fichier : MOG8-13.xlsx Feuilles :  SéanceT et Fig1 

 
Cellule Formule Copiée dans : 

B5 =SI(OU(M4=NbP;M4="");"";B4+LMin+(LMax-LMin)*ALEA()) B6:B53 

C4 * =SI(B4="";"";ALEA()) C5:C53 * 

E5 =SI(B5="";"";MAX(B5;SI(E4>B4;E4;L4+D5))) E5:E53 

F4 * =SI(OU(B4="";E4>B4);"";ALEA()) F5:F53 * 

H4 =SI(F4="";"";SI(ALEA()<=p;1;0)) H5:H53 

I4 * =SI(OU(B4="";E4>B4;H4=1);"";ALEA()) I5:I53 * 

L4 =SI(B4="";"";MAX(E4;G4;J4)) L5:L53 

M4 =SI(B4="";"";M3+MAX(H4;K4)) M5:M53 

* Cellules masquées 

 

 Les données des colonnes B, E, G, J et L représentent des temps (en minutes) écoulés depuis 

le début de la séance qui, selon l'approche retenue dans le fichier MOG8-13.xlsx, est le 

moment où le premier pigeon est lancé. 

 

 La colonne B indique quand s'effectuent les différents lancers.  Le premier coïncide toujours 

avec l'instant t = 0 auquel nous avons convenu de démarrer l'horloge et il n'est pas nécessaire 

de recourir à une formule en B4.  Pour les autres, on additionnera au temps du lancer 

précédent une valeur pseudo-aléatoire de la variable L.  Comme celle-ce obéit à une loi 

uniforme, l'ajout requis se traduit par « +LMin+(LMax-LMin)*ALEA() ». 
 

 En colonne E, nous déterminons quand le tireur est prêt à entamer le processus de viser et de 

tirer.  Trois situations sont traitées par la formule de la colonne E. 

o Considérons d'abord le 9
e
 lancer qui est décrit sur la ligne 12 : l'essai précédent 

s'est terminé en t = 67,8 (voir L11) et le tireur a rechargé son arme en 5,8 secondes 

(voir D12); en t = 67,8 + 5,9 =73,6, il attend le prochain lancer, qui est effectué en 

t = 74,0 (voir B12).  Il commence alors à viser.  Ainsi, c'est la valeur de B12 qui 

est renvoyée par la formule en E12. 

o Le 6
e
 essai se termine en t = 50,3 (voir L9).  Le tireur prend 6,4 secondes pour 

recharger (voir D10); il est donc prêt en t = 50,3 + 6,4 = 56,7.  Mais le 7
e
 pigeon a 

déjà été lancé (en t = 54,6 : voir B10).  Il ne tirera donc pas lors de cet essai. La 

valeur 56,7 affichée en E10 provient donc de l'expression « L9+D10 ». 

o Il arrive parfois que deux pigeons soient lancés pendant que le tireur recharge son 

arme. Cette situation survient plutôt rarement – 52 fois dans l’échantillon de 1000 

séances considéré dans les figures 2 et 3.  Le 3
e
 essai se termine en t = 24,6 (voir 

L6), le tireur prend 8,9 secondes pour recharger (voir D7) et est prêt à viser en t = 

33,5 (voir E7); mais le 4
e
 pigeon est lancé en t = 24,5 et le 5

e
, en t = 33,2 (voir 

B8); la valeur renvoyée en E8 doit donc coïncider avec celle de E7.  Dans une telle 

situation où deux lancers sont effectués pendant que l'arme est rechargée, la valeur 

en colonne E sur la ligne associée au second lancer est égale à celle dans la même 

colonne sur la ligne précédente. 
 

 Les colonnes F, G et H décrivent les premiers tirs des différentes séances.  Considérons 

d’abord la formule de F, illustrée dans la figure 1 par sa version pour F4. Évidemment, il n'y 

a pas de tir si la séance est terminée (B4="") ou si le tireur n'a pas fini de recharger au moment 

du lancer (E4>B4).  Dans ces cas, les cellules des colonnes F. G et H restent vides.  Sinon, le 

tireur vise le pigeon, ce qui se traduit dans notre modèle de simulation par une valeur 

pseudo-aléatoire de la variable V qui  se calcule selon la formule utilisée dans la figure 8.19 ; 

le temps V requis pour viser s'ajoute à la valeur en colonne E, pour donner le moment affiché 

en colonne G où s'effectue le 1
er

 tir. 
 

 Pour simuler le résultat du 1
er

 tir, nous générons un nombre pseudo-aléatoire entre 0 et 1 et le 

comparons à la probabilité p = 0,8 que le tireur atteigne un pigeon.  Si la valeur renvoyée par 

ALEA() est inférieure ou égale à p, nous considérons que le tir est réussi et 1 sera affiché. 
 

 Les colonnes I, J et K décrivent le 2
e
 tir et les formules utilisées sont semblables à celles des 

colonnes F, G et H.  D’ailleurs, celles de J4 et de K4 ont été obtenues en copiant le contenu de 

la plage G4:H4. Il faut cependant s'assurer que le 2
e
 tir n'est pas effectué quand le pigeon est 

atteint lors du 1
er
; c'est pourquoi la condition « H4=1 » a été ajoutée dans la formule de I4. 

 

 La fin de l'essai numéro 1 survient au plus tardif des trois moments suivants : quand le tireur 

est prêt (E4), quand il effectue le 1
er

 tir (G4) ou le second (J4).  D'où le recours à la fonction 

Max dans la formule de L4. 
 

 Enfin, le nombre cumulé de succès s'obtient en ajoutant le résultat de l'essai courant au total 

sur la ligne précédente.  Noter que la formule « MAX(H4;K4) » donne 0 quand les cellules H4 

et K4 sont toutes deux vides. 

 

Il faut maintenant simuler 1000 séances.  Dans la figure 2, nous avons synthétisé en ligne 3 les 

résultats de la feuille SéanceT, puis nous avons utilisé la commande Table de données… d'Excel.  

La formule en B3, dont l'objectif annoncé est de compter le nombre de lancers lors d'une séance, 

indique en fait combien il y a de cellules non vides parmi les 50 de la plage B4:B53 de la feuille 

SéanceT.  Noter que, dans la séance simulée reproduite à la figure 1, les cellules B43 à B53 sont 

vides et que cette séance a exigé 50 – (53 – 42) = 39 essais.  De même, le nombre de tirs 

effectués s'obtient comme le nombre de cellules non vides au total dans les plages G4:G53 et 

J4:J53.  Enfin, la durée d'une séance est le temps à notre horloge lors de la fin du dernier essai; et 

ce temps est toujours la valeur maximale apparaissant en colonne L. 

Fichier : MOG8-13.xlsx Feuilles :  Simul et Fig2 

 
Cellule Formule 

B3 =50-NB.VIDE(SéanceT!B4:B53) 

C3 =100-NB.VIDE(SéanceT!G4:G53)-NB.VIDE(SéanceT!J4:J53) 

D3 =MAX(SéanceT!L4:L53) 

 

La figure 3 résume les résultats de notre simulation.  Le tireur met en moyenne 346,2 secondes 

pour abattre 20 pigeons, mais une séance peut se prolonger jusqu'à 432,3 secondes, soit un peu 

plus de 7 minutes et 12 secondes.  En moyenne, 35,3 pigeons sont lancés et il tire 26,3 fois.  Dans 

un cas, 43 pigeons ont été nécessaires (voir C7).  Une autre fois, il lui a fallu 35 tirs.  Noter que 

ces deux situations extrêmes ne sont pas produites en même temps : lors de l’unique séance où le 

maximum de 35 tirs a été atteint (n
o
 138, ligne 141 de la feuille Fig2), seulement 40 pigeons ont 

été lancés. 

Fichier : MOG8-13.xlsx Feuilles :  Résultats et Fig3 

14. Les rendez-vous de l’ophtalmologiste 

 

(a) Dans un premier temps, nous analysons une journée typique.  La figure 1 indique 

comment nous avons procédé.  Les données des colonnes C, F et I représentent le temps (en 

minutes) écoulé depuis 8h15, le début de la journée de travail de la spécialiste : par exemple, le 

ligne 11 (numéro 8) correspond à la période 10h00-10h15 qui commence (8 – 1) × 15 = 105 

minutes après 8h15; en principe, le patient arrive 10 minutes plus tôt, à 9h50, soit 105 – 10 = 95 

minutes après le démarrage de notre «horloge» et c'est pourquoi nous avons reporté la valeur 95 

en C11.  Le dernier rendez-vous est fixé à 12h45, soit 4h30 après le tout premier qui est à 8h15; 

par conséquent, une journée de travail du médecin comprend un maximum de 1 + (4 × 4) + 2 = 

19 consultations. 

 

Fichier : MOG8-14a.xlsx Feuilles :  JourT et Fig1 

 
Cellule Formule Copiée dans :  

B4 =SI(ALEA()<=p;1;0) B5:B22 

C4 =SI(B4=0;"";15*A4-25) C5:C22 

D4 =SI(B4=0;"";(I3>290)+0) D5:D22 

E4 =SI(OU(B4=0;D4=1);"";F4-C4) E5:E22 

F4 =SI(OU(B4=0;D4=1);"";MAX(I3;C4)) F5:F22 

G4 =SI(OU(B4=0;D4=1); "";ALEA()) G5:G22 

I4 =SI(OU(B4=0;D4=1);I3;F4+H4) I5:I22 

 

Considérons, à titre d'exemple, la ligne 4, qui correspond au 1
er

 patient.  Nous traiterons les 

colonnes dans l'ordre B, C, D, G, H, I, E et F. 
 

 En B4, nous testons si ce dernier se présente.  Il s'agit de générer un nombre pseudo-aléatoire 

entre 0 et 1, et de déclarer que le patient vient au rendez-vous quand ce nombre est inférieur 

ou égal à p = 0,9.  Dans le cas négatif, la consultation ne pourra évidemment avoir lieu et la 

plage C4:H4 restera vide; les formules des différentes cellules de cette plage incorporent 

donc un test qui renvoie une cellule vide lorsque B4 contient 0. 
 

 En C4, le test mentionné ci-dessus prend la forme « =SI(B4=0;"";…) ».  Si B4  0, le temps 

d'arrivée du 1
er

 patient se calcule ainsi : (15 × 1) – 25 = –10.  Ce temps d'arrivée est 

augmenté de 15 minutes lorsqu'on passe d'une ligne à la suivante, ce qui traduit bien la 

politique de la spécialiste d'espacer ses rendez-vous d'un quart-d'heure. 
 

 Toujours dans le contexte où B4  0, il faut ensuite tester si le patient est renvoyé ou non.  

On pourrait prétendre que les premiers patients seront sûrement acceptés et qu'effectuer le 

test seulement dans les dernières lignes suffirait amplement.  Mais, où placer la frontière, à 

quelle période démarrer la procédure de test?  Pour simplifier, et aussi pour que les formules 

dans une colonne soient toutes semblables et s'obtiennent en copiant celles de la ligne 4, nous 

convenons de faire le test systématiquement pour tous les patients.  Il s'agit en fait de vérifier 

si le temps à l'horloge I sur la ligne précédente, qui représente le moment où l’ophtal-

mologiste sera disponible pour rencontrer ce patient, est supérieur à 290 (à 13h05, le moment 

à partir duquel les gens sont renvoyés, il s'est écoulé 290 minutes depuis le démarrage de 

notre horloge à 8h15).  Dans la formule en D4, nous avons ajouté le terme « + 0 » afin que la 

conclusion du test s'affiche sous la forme 1 ou 0, et non vrai ou faux. 
 

 Si le patient est renvoyé (1 en colonne D), le reste de la ligne est vide, à l'exception de la 

colonne I.  Sinon, il nous faudra simuler la consultation.  C'est pourquoi les formules des 

colonnes E à G commencent par la fonction SI d'Excel : 

=SI(OU(B4=0;D4=1);"";…) 

Si le client ne se présente pas (B4 = 0) ou est refusé (D4 = 1), la formule renvoie une cellule 

vide.  Sinon, un certain calcul sera fait… 
 

 Nous nous intéressons maintenant à la consultation proprement dite.  Pour obtenir sa durée, 

nous générons en colonne G un nombre pseudo-aléatoire entre 0 et 1, que nous transformons 

en colonne H en une valeur d'une loi triangulaire tronquée à l'aide d’une version adaptée au 

présent contexte de celle utilisée en section 8.2.4 (voir figure 8.19, colonne G). Voici la 

formule qui apparaît en H4 et qui sera copiée dans la plage G5:G22 : 
 

MIN(20;SI(G4<=Xk;XMin+(XMax-XMin)*RACINE(Xk*G4);XMax-(XMax-XMin)*RACINE((1-Xk)*(1-G4)))) 
 

où Xk = (XMod ‒ XMin) / (XMax ‒ XMin) = (15 ‒ 12) / (28 ‒ 12) = 0,1875. 
 

 La colonne I contient une «horloge» indiquant quand la spécialiste sera disponible pour le 

prochain patient : le temps reporté sur une ligne correspond au moment où se termine la 

consultation quand il y en a une, et au temps affiché en I sur la ligne précédente dans le cas 

contraire; évidemment, le temps de fin d'une consultation est égal à la somme du temps de 

début (colonne F) et de la durée (colonne H). Illustrons par deux exemples tirés de la figure 1 

comment se calcule le moment où la spécialiste sera disponible pour le prochain patient : 

o ligne 4 : la formule en I4 retourne la valeur 17,9 qui est la somme du temps (F4 = 0) 

où débute la consultation du 1
er
 patient et de la durée (H4 = 17,9) de cette consultation; 

o ligne 5 : comme le 2
e
 patient ne s'est pas présenté, le moment où la spécialiste 

devient disponible reste le même que sur la ligne précédente, soit 17,9. 
 

 L'attente d'un patient (colonne E) est l'écart entre son arrivée (colonne C) et le début de sa 

consultation (colonne F).   

 Il nous reste à indiquer comment obtenir les valeurs en colonne F.  Deux cas peuvent 

survenir. 

o Si le médecin est disponible lorsque le patient arrive, le début de la consultation 

coïncide avec l'heure d'arrivée du patient.  Par exemple, le 3
e
 patient arrive à 

l'instant t = 20 (voir C6); commme la spécialiste est libre depuis l'instant t = 17, 9 

(voir I5), elle rencontrera le patient sans délai et c'est la valeur t = 20 affichée en 

C6 qui est reportée en F6. 

o Par contre, le 4
e
 patient arrive à l'instant t = 35 (voir C7) et devra attendre que le 

médecin se libère; cette fois c'est la valeur t = 36,1 affichée en I6 qui est reportée en F7. 

 

Nous avons résumé les résultats de la figure 1 dans la ligne 3 de 1a feuille Fig2, puis nous avons 

généré 1000 journées à l'aide de la commande Table de données… d'Excel. Les résultats obtenus 

sont reproduits dans le figure 2 ci-dessous.   

 

Fichier : MOG8-14a.xlsx Feuilles :  Simul et Fig2 

 
Cellule Formule 

B3 =19-NB.VIDE(JourT!H4:H22) 

C3 =SOMME(JourT!D4:D22) 

D3 =MOYENNE(JourT!E4:E22) 

E3 =JourT!E11 

F3 =JourT!E19 

 

Enfin, la figure 3 (voir page suivante) donne les principales statistiques de notre échantillon de 

1000 journées.  La spécialiste traite en moyenne 16,5 personnes par jour et celles-ci attendent 

18,8 minutes en moyenne avant le début de leur consultation.  Cependant, pour plus de 5% des 

journées, l'attente moyenne excède une demi-heure (voir D6).  On peut comprendre les 

récriminations des plus impatients des patients… et l'intérêt pour la clinique de réviser cette 

politique de rendez-vous. 

Fichier : MOG8-14a.xlsx Feuilles :  Résultats et Fig3 

 
 

 

(b) Nous reprenons l'approche appliquée dans la question précédente, en apportant les légers 

ajustement requis.  Ainsi, la feuille Fig4, où est analysée une journée typique, comprend 

maintenant 15 rendez-vous.  De plus, les formules des colonnes C, D et H sont modifiées pour 

tenir compte du changement de politique : en C4, le coefficient de A4 est 20, et non pas 15, 

puisque l'intervalle entre deux rendez-vous serait de 20 minutes; le point de comparaison en D4 

pour renvoyer un patient est 1h00, soit 300 minutes après le démarrage de notre horloge à 8h00; 

enfin, la durée de la consultation n'est plus limités à 20 minutes et la formule de la colonne H ne 

fait plus intervenir la fonction Min d'Excel.  

 

Fichier : MOG8-14b.xlsx Feuilles :  JourT et Fig4 

 
Cellule Formule Copiée dans :  

C4 =SI(B4=0;"";20*A4-30) C5:C22 

D4 =SI(B4=0;"";(I3>300)+0) D5:D22 

La figure 5 ci-dessous résume les résultats obtenus.  On constate que, dans notre échantillon, 

personne ne serait renvoyé; que la spécialiste traiterait en moyenne 13,3 patients par jour; que 

ceux-ci attendraient 3,5 minutes en moyenne. 

 

Fichier : MOG8-14b.xlsx Feuilles :  Résultats et Fig5 

 
 

 

 

15. Coïncidence d'anniversaires dans un groupe 

 

(a) La figure 1 (voir page suivante) indique comment nous avons procédé pour tester si, 

parmi n personnes choisies au hasard, au moins deux ont la même date d'anniversaire.  La taille n 

du groupe considéré est représentée dans notre fichier MOG8-15a.xlsx par le paramètre TailleG. 

 

Les trois formules commencent par un test, car nous voulions utiliser la même feuille pour traiter 

des groupes de 20 à 25 personnes.  Le test renvoie une cellule vide quand le numéro en colonne A 

dépasse la taille du groupe.  Ainsi, dans la figure 1 où le paramètre TailleG a été fixé à la valeur 20, 

les 5 dernières lignes n'affichent que le numéro, tandis que les 20 premières simulent chacune un 

individu différent. 

 

La formule de la colonne B renvoie un entier entre 1 et 365 quand la condition du test est 

satisfaite.  En effet, « 365*ALEA() » génère un nombre réel pseudo-aléatoire entre 0 et 365, cette 

dernière valeur étant exclue; la fonction ENT arrondissant à l'entier immédiatement inférieur, 

« ENT(365*ALEA()) » retourne un nombre entier entre 0 et 364; le résultat final, après le décalage 

dû au terme « +1 », se situe entre 1 et 365 tel que désiré.  De plus, les 365 valeurs possibles sont 

équiprobables. 

 

La fonction RANG(B3; B$3:B$27;1) commande à Excel de trier en ordre croissant les nombres 

apparaissant dans la plage B3:B27, puis de renvoyer le rang de celui en B3.  Par exemple, la plus 

petite valeur dans la colonne B de la figure 1 est 39 (voir cellule B20) et on retrouve le rang 1 

dans la cellule C20; de même, 44 vient au 2
e
 rang (voir ligne 7).  Un nombre qui revient plusieurs 

fois dans la plage se voit attribuer le même rang lors de toutes ses occurrences : ainsi, 94 apparaît 

deux fois (voir B5 et B10) et le rang 4 est reporté dans les cellules C5 et C10; noter que 107, la 

valeur suivante de la liste, vient au 6
e
 rang, et non au 5

e
 (voir C19). 

Enfin, la formule NB.SI(C$3:C$27;A3) compte le nombre d'occurrences dans la plage C3:C27 de la 

valeur en A3.  Ainsi, le rang 1 apparaît une seule fois en colonne C et Excel renvoie 1 en D3.  Par 

contre, le rang 4 est partagé par deux journées de notre échantillon (voir lignes 5 et 10) et la 

valeur 2 est affichée en D6. 

 

Fichier : MOG8-15a.xlsx Feuilles :  GroupeT et Fig1 

 
Cellule Formule Copiée dans :  

B3 =SI(A3<=TailleG;1+ENT(365*ALEA());"") B4:B27 

C3 =SI(A3<=TailleG;RANG(B3;B$3:B$27;1);"") C4:C27 

D3 =SI(A3<=TailleG;NB.SI(C$3:C$27;A3);"") D4:D27 

 

La cellule A5 de la figure 2 (voir page suivante) indique une façon de tester si au moins deux 

dates d’anniversaire coïncident dans un groupe donné : il suffit en effet de vérifier si on retrouve 

un nombre supérieur à 1 dans la colonne D.  Pour le groupe représenté à la figure 1, la présence 

de la valeur 2 dans la cellule D6 de permet de conclure que les 2 personnes qui occupent le rang 4 

dans ce groupe partagent la même date d’anniversaire.  Par contre, si les n individus du groupe 

ont tous des dates d’anniversaire différentes, la colonne D de la feuille GroupeT contiendra 

seulement des « 1 ». 

 

Fichier : MOG8-15a.xlsx Feuilles :  Simul et Fig2 

 
Cellule Formule  

A5 =SI(MAX(GroupeT!D$3:D$27)>1;1;0)  

 

La simulation de la figure 2 a été produite à à l'aide de la commande Table de données… d'Excel : 

nous avons d'abord sélectionné la plage A5:G5005; puis, nous avons défini de la façon suivante 

les paramètres de cette commande : 
 

 Cellule d'entrée en ligne: $E$1  (cette cellule contient la valeur du paramètre TailleG qui 

représente la taille n du groupe). 
 

 Cellule d'entrée en colonne: $A$5. 

 

La tableau qui suit résume les résultats obtenus. On constate que, dans nos échantillons, la 

proportion des groupes où se produit au moins une coïncidence dépasse 50% quand celui-ci 

comprend 23 personnes ou plus.  Ce qui confirme l'affirmation de l'énoncé. 

 

Taille 20 21 22 23 24 25 

Proportion 0,4070 0,4420 0,4778 0,5144 0,5360 0,5662 

 

 
Note.  Voici comment calculer de façon exacte la probabilité pn qu'il y ait au moins une coïncidence dans 

un groupe de taille n, sous l'hypothèse que les gens sont choisis de façon aléatoire et que les 365 jours de 

l'année sont équiprobables : 

1 – pn  =  P(n dates différentes) = 
(

365
𝑛

)×𝑛!

365𝑛  =  
365! × 𝑛!

365𝑛  × 𝑛! × (365−𝑛)!
 =   

365! 

365𝑛  × (365−𝑛)!
 . 

Le tableau ci-dessous donne les valeurs de pn.  L'écart entre la probabilité pn et l'estimé fourni par 

l'échantillon est inférieur à 0,5% dans tous les cas, sauf lorsque n = 23 et alors l'écart est d'environ 0,7%. 

 

Taille 20 21 22 23 24 25 

pn 0,4114 0,4437 0,4757 0,5073 0,5383 0,5687 

(b) Il suffit de répéter les calculs effectués dans la question précédente, mais en fixant la taille 

du groupe aux valeurs 35, 36, 40 et 41.  Voici les résultats de la simulation. 

Taille 35 36 40 41 

Proportion 0,819 0,830 0,882 0,898 

Le professeur propose aux étudiants un pari qui est fortement biaisé en sa faveur.  On pourrait 

s'interroger sur son sens de l'éthique… à moins que les sommes en jeu soient si faibles que le côté 

pédagogique prime sur l'aspect financier. 

 

(c) Il suffit de répéter les calculs effectués dans la question (a), mais en faisant varier la taille 

du groupe de 85 à 90 et en modifiant ainsi la formule de la cellule A5 de la feuille Simul :   

=SI(MAX(GroupeT!D$3:D$92)>2;1;0) 

Voici les résultats de la simulation. 

Taille 85 86 87 88 89 90 

Proportion 0,4515 0,4790 0,4900 0,5075 0,5145 0,5310 

 

 

 

16. Les machines à sous. 

 

(a) Certaines plages de la feuille Données, auxquelles nous référons dans les formules 

apparaissant dans les autres feuilles, se sont vus attribuer des noms : 

 dans les simulations, le numéro de la facette d'un disque sera noté X; la liste des valeurs de 

cette variable aléatoire, qui occupent les cellules B11 à B17, sera nommée ValeursX; 

 la plage B11:C17 donne les prix (en $) lorsque les trois disques affichent le même symbole 

et porte le nom Prix3; 

 Cumul1, Cumul2 et Cumul3 réfèrent aux plages E11:E17, G11:G17 et I11:I17 respectivement. 
 

Fichier : MOG8-16.xlsx Feuilles :  Données et Fig1 

 
Cellule Formule Copiée dans :  

D18 =SOMME(D11:D17) F18, H18 

E12 =E11+D11/D$18 E13:E17, G13:G17, I13:I17 

Dans un premier temps, nous simulons la séance d'un joueur typique.  La figure 2 qui suit indique 

comment nous avons procédé.  Chacune des lignes 3 à 202 représente l'un des 200 coups.  Les 

colonnes B, C et D indiquent quels symboles apparaissaient sur chacun des trois disques : par 

exemple, la formule en B3 commande de générer un nombre pseudo-aléatoire entre 0 et 1, puis 

de le transformer en un numéro de 1 à 7 en utilisant les fréquences cumulées de la plage E11:E17 

de la feuille Données.  Ainsi, dans le cas illustré à la figure 2, la fonction ALEA() a renvoyé un 

nombre entre 0,545 et 0,636; cette valeur correspondant au numéro 6, le disque 1 affichait une 

cerise lors de ce 1
er

 coup.  La formule de la colonne E donne le nombre de cerises pour le coup 

associé à la ligne; plus techniquement, il s'agit de compter combien de fois apparaît le numéro 6 

dans les colonnes B, C et D de cette ligne. 

 

Enfin, la cellule F3 calcule le prix remporté par le joueur lors de ce coup à l'aide de la formule : 

=SI(E3<>0;CHOISIR(E3;2;5;20);SI(ET(B3=C3;C3=D3;B3<>7);RECHERCHEV(B3;Prix3;2;0);"")) 

Deux cas sont considérés. 

 Si au moins une cerise a été obtenue, autrement dit si la valeur en colonne E de cette ligne est 

non nulle, on choisit entre des prix de 2$, 5$ ou 20$ selon que le nombre de cerises est 1, 2 

ou 3. 

 Sinon, on teste si les trois disques affichent le même symbole.  Dans le cas négatif, le joueur 

ne gagne pas ce coup-là et on ne reporte rien en colonne F.  Dans le cas contraire, la formule 

renvoie le montant du prix, pourvu que le symbole commun ne soit pas un citron.  Noter 

qu'aucun prix n'est accordé quand le joueur obtient trois citrons et que la formule ne renvoie 

rien dans ce cas. 

 

Fichier : MOG8-16.xlsx Feuilles :  JoueurT et Fig2 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Cellule Formule Copiée dans :  

B3 =RECHERCHE(ALEA();Cumul1;ValeursX) B4:B202 

E3 =(B3=6)+(C3=6)+(D3=6) E4:E202 

F3 =SI(E3<>0;CHOISIR(E3;2;5;20);…) F4:F202 

 

La ligne 3 de la figure 3 (voir page suivante) résume les résultats de la séance illustrée à la 

figure 2.  La formule de B3 compte le nombre de prix récoltés par le joueur en 200 coups : en 

effet, le fait de ne pas gagner lors d'un coup donné se traduit par une cellule vide dans la 

colonne F; le nombre de gains pendant une séance de 200 coups s'obtient donc en retranchant de 

200 le nombre de cellules vides de la plage F3:F202 de la feuille JoueurT.  La cellule C3 

s'intéresse, elle, à la valeur des prix : plus précisément, elle calcule la somme des prix reçus par le 

joueur pendant la séance.  Enfin, la cellule D3 donne le revenu net du joueur : celui-ci s'obtient en 

déduisant de son revenu (C3) la somme investie, qui est égale au produit du nombre de coups 

(200) par la mise pour un coup (1$). 

Fichier : MOG8-16.xlsx Feuilles :  SimulA et Fig3 

 

 

 

 

 

 

 

 

 

 

 

 

Cellule Formule   

B3 =200-NB.VIDE(JoueurT!F3:F202)  

C3 =SOMME(JoueurT!F3:F202)  

D3 =C3-200*1  

 

La séance typique est ensuite répétée 1000 fois à l'aide de la commande Table de données… 

d'Excel.  Les résultats obtenus sont reproduits dans le figure suivante. On note, sans surprise, que 

la machine mérite son surnom de «bandit manchot».  En effet, dans l'échantillon considéré ici, le 

revenu net médian (D3) s'établit à -31$; de plus, les joueurs perdent 15,20$ en moyenne (D8).  

Cependant, quelques chanceux réussissent à sortir gagnants de la séance: par exemple, l'un des 

1000 joueurs a complété la séance avec des gains nets de 539$. 

Fichier : MOG8-16.xlsx Feuilles :  RésultA et Fig4 

 

 

 

 

 

 

 

 

 

 

 

 

(b) Les pertes des joueurs se traduisent en gains pour le propriétaire : ainsi, pour l'échantillon 

illustré considéré à la question (a), le revenu net moyen du propriétaire est de  15,20$ par joueur. 

(c) La figure 5 illustre une soirée où la séance de Charles a duré 640 coups.  Pour le 1
er

 coup 

(ligne 3), nous avons copié les formules de la figure 2 dans la plage B3:F3.  Mais, à partir du 2
e
 

coup, nous avons ajouté un test pour vérifier si la séance s'est poursuivie jusque-là.  Plus 

techniquement, il s'agit de tester la valeur du solde sur la ligne précédente : s'il est nul, la séance 

est terminée et les formules des colonnes B à F renvoient des cellules vides; sinon, Charles 

continue à jouer et nous reprenons les formules de la ligne 3. 

 

Deux cas sont à considérer pour le calcul du solde de la colonne G. 

 Une fois que le solde est tombé à 0$, la séance est terminée et le solde sur les lignes 

subséquentes sera nul. 

 En cours de séance, le solde s'obtient en retranchant du solde précédent la mise de 1$, puis 

en ajoutant le prix éventuel reporté en colonne F.  (Comme Excel renvoie un message 

d'erreur quand on lui commande d'additionner une cellule numérique et une cellule vide, 

nous avons inclus un test sur le contenu de la cellule en colonne F : si celle-ci est vide, le 

joueur n'a pas reçu de prix lors de ce coup et le terme représentant le prix sera le nombre 0.) 

 

Fichier : MOG8-16.xlsx Feuilles :  SoirT et Fig5 

 
Cellule Formule Copiée dans :  

B4 =SI(G3=0;"";RECHERCHE(ALEA();Cumul1;ValeursX)) B5:B1002 

E4 =SI(G3=0;"";(B4=6)+(C4=6)+(D4=6)) E5:E1002 

F4 =SI(G3=0;"";SI(E4<>0;CHOISIR(E4;2;5;20);…)) F5:F1002 

G4 =SI(G3=0;0;G3-1+SI(F4="";0;+F4)) G5:G1002 

 

Nous avons réservé 2000 lignes pour les séances.  Cette valeur est parfois insuffisante : en effet, 

comme le montre la figure 6 (voir page suivante), lors de la soirée numéro 997, Charles possédait 

306$ après 2000 coups.  L'allongement considérable de certaines séances provient des prix de 

500$ qui, chacun, assurent au joueur 500 coups additionnels.   

 

Noter que la formule de B3 indique combien de coups ont été nécessaires ce soir-là pour épuiser 

les 20$ investis; que celle de C3 donne le solde final entre les mains de Charles après 2000 coups 

lorsque la séance dépasse cette limite. 

Fichier : MOG8-16.xlsx Feuilles :  SimulC et Fig6 

 
Cellule Formule   

B3 =2000-NB.SI(SoirT!G3:G2002;0)+1  

C3 =SoirT!G2002  

 

La figure 7 résume les résultats de la simulation de la figure précédente.  En moyenne, Charles 

aurait joué 173 coups par soir (B13).  Mais, à 2 reprises, il n'aurait gagné aucun prix (B2) et sa 

séance se serait terminée dès le 20
e
 coup.  Par contre, 22 séances se seraient prolongées au-delà 

du 2000
e
 coup (B4).  Si Charles avait dû cesser après ce 2000

e
 coup – par exemple, parce que le 

bar fermait à ce moment-là –, il aurait dépensé en moyenne 20 – 9,41 = 10,59 dollars (B5); un 

soir, il aurait même quitté avec 903$ en poche (B6). 

 

Fichier : MOG8-16.xlsx Feuilles :  RésultC et Fig7 

Note.  On peut montrer théoriquement (voir fichier MOG8-16.xlsx, feuille Prob) que la perte attendue 

après 200 coups est de 18,18$.  De plus, la probabilité que Charles ne gagne aucun prix lors des 20 

premiers coups est de 0,09%; on en déduit que la variable S, «nombre de soirs dans un échantillon de 1000 

qui se terminent dès le 20e coup», obéit à une loi binomiale B(1000; 0,09%) et que E(S) = 0,9.  Enfin, la 

distribution de S est résumée dans le tableau suivant. 

 

s 0 1 2 3 4 5 6+ 

P(S = s) 39,92% 36,68% 16,83% 5,14% 1,18% 0,22% 0,04% 

 

 

 

 

17.  Rupture de stock et rabais 

 

Nous décrivons ici seulement les modifications à apporter au modèle de gestion des stocks décrit 

dans la section 8.2.3.  La feuille Données contient deux paramètres, en plus de ceux énumérés au 

tableau 8.6 : 

 CRabais = 50 représente le montant (en $) offert par le magasin en cas de rupture de stock. 

 PropRev = 70% est la proportion des clients qui acceptent de revenir lorsqu'on leur propose 

le rabais. 

 

La figure 1 qui suit reproduit quelques extraits de la feuille Simul où sont calculés les coûts pour 

une période de 5000 jours consécutifs. 

 

Fichier : MOG8-17.xlsx Feuilles :  Simul et Fig1 

 
Cellule Formule Copiée dans :  

C7 =TComm*NB.SI(L3:L6;A7) C8:C5002 

D7 =B7+C7-SI(C7=0;0;J6) D4:D5002 

I7 =CRITERE.LOI.BINOMIALE(H7;PropRev;ALEA()) I4:I5002 

J7 =SI(C7=0;J6+I7;0) J4:J5002 

M7 =M6-F7+SI(K7="Oui";TComm;0)-I7 M4:M5002 

N7 =CStock*(D7+G7)/2+CPénurie*(H7-I7)+CComm*(K7="Oui") N4:N5002 

O7 =MargeU*F7-N7+(MargeU-CRabais)*I7 O4:O5002 

 

Les formules des colonnes B, E, F et G sont identiques à celles du fichier Gestion-Stocks.xlsx (voir 

figure 8.16) et ne sont pas reproduites ici.  Celles de I et J calculent le nombre de rabais acceptés 

par les clients et sont commentées ci-dessous.  L'inclusion de ces deux colonnes force le décalage 

des colonnes subséquentes, ce qui entraîne divers ajustements à plusieurs formules de la figure 

8.16; dans certains cas – C7, K7 et L7–, la seule différence entre les formules des fichiers 

Gestion-Stocks.xlsx et MOG8-17.xlsx réside dans les noms des colonnes auxquelles elles réfèrent; à 

titre d'exemple, nous donnons la version modifiée de celle de C7. 

 

Nous commentons maintenant les changements dans la feuille Simul qui découlent de l'hypothèse 

qu'en cas de rupture de stock, une proportion des clients acceptent de revenir acheter la batterie 

en échange d'un rabais. 

 Pour simplifier, nous considérons que tous les clients en possession d’un «coupon-rabais» de 

50 $ se présenteront le jour de la réception d'une commande.  Nous convenons donc de 

modifier de la façon suivante la formule du stock à l’ouverture (D7) : dans la figure 8.16, la 

valeur en D7 se calcule comme la somme des unités restant la veille au soir (G6, repris en 

B7) et celles reçues le matin même (C7); ici, de ce total, nous retranchons le nombre de 

coupons-rabais en circulation (J6) quand une commande est reçue (C7  0). 

 Le nombre C de coupons-rabais émis un jour donné (I7) obéit à une loi binomiale dont le 

paramètre «nombre de tirages» est égal au nombre d'unités en pénurie reporté en H7.  La 

fonction ALEA() de la formule en I7 commande de générer un nombre pseudo-aléatoire entre 

0 et 1, que la fonction CRITERE.LOI.BINOMIALE transforme en une valeur de la variable C. 

 Nous avons supposé que tous les clients profitent de leurs coupons-rabais le plus tôt possible.  

Par conséquent, si une commande est reçue un matin donné, il ne restera aucun coupon en 

circulation ce soir-là. Plus techniquement, il s'agit d'exiger que J7 soit nul quand C7  0.  

Dans le cas contraire, le nombre cumulé de coupons en circulation un soir donné (J7) est la 

somme du nombre de coupons en circulation la veille au soir (J6) et du nombre de coupons 

émis ce jour-là (I7). 

 Le stock au livre (M7) inclut les unités réellement en stock, plus les unités commandées et en 

instance de livraison, moins les unités en pénurie pour lesquelles le magasin a émis un 

coupon.  Nous ajoutons donc le terme « –I7 » à la formule utilisée dans le modèle de la 

section 8.2.3. 

 Dans les coûts de gestion, le coût de pénurie s'applique seulement aux clients qui refusent le 

coupon-rabais de 50$.  C'est pourquoi nous multiplions ici le paramètre CPénurie par (H7–I7). 

 Lorsqu’il y a pénurie et que le client accepte de revenir plus tard chercher sa batterie, les 

revenus de la vente reportée dans le temps sont imputés à la journée où le client est venu 

pour la 1
re

 fois et a pris la décision d’achat.  C'est ce qui explique la présence du terme 

« (MargeU–CRabais)*I7 » dans la formule de la cellule O7. 

 

Note. Comme dans le modèle de la section 8.2.3, la cellule B4 et la plage D4:O4 sont copiées dans les 

lignes 5 à 5002.  Mais, c’est seulement à partir de la ligne 7 que la formule de la colonne C est «stable» et 

peut être copiée. 

 

Comme dans la section 8.2.3, nous poursuivons notre étude en analysant l'impact des paramètres 

TComm et PComm à l'aide de la commande Table de données… d'Excel. Les résultats obtenus 

sont reproduits dans le tableau ci-dessous.  On observe, comme dans la section 8.2.3, que les 

variations de revenu sont, à quelques exceptions près, peu ou pas significatives, pourvu que 

PComm soit supérieur à 20. 

 

    Nombre de Nombre total d'unités Demande Revenu 

TComm PComm commandes Pénurie Stock totale net moyen 

100 10 381 3 037 223 312 38 979 2 043 

  15 381 1 699 243 313 38 677 2 056 

  20 386 1 022 264 903 38 938 2 087 

  25 386 572 288 167 38 725 2 085 

  30 385 173 316 659 38 590 2 087 

  35 390 48 342 126 38 998 2 111 

  40 388 43 357 305 38 756 2 097 

  45 388 1 383 445 38 747 2 096 

  50 387 2 412 636 38 680 2 091 

100 20 388 976 264 113 39 047 2 095 

150   257 672 390 270 38 762 2 085 

200   192 482 513 196 38 532 2 073 

250   154 324 643 149 38 553 2 075 

300   130 377 765 471 38 929 2 088 

350   111 230 886 460 38 933 2 087 

400   97 216 1 014 141 38 906 2 080 

450   86 172 1 140 112 38 805 2 069 

500   79 330 1 251 045 39 172 2 080 

 

 

 

18. Les quatre relais d’un panneau de contrôle. 

 

La figure 1 (voir page suivante) analyse 1051 pannes consécutives dans le contexte où la 

politique de l'entreprise consiste à remplacer seulement le relai défaillant.  On notera que les 

formules en B3 et en B4 diffèrent.  En effet, au départ (B3:E3), les quatre relais sont neufs et leur 

durée de vie s'obtient en générant une valeur pseudo-aléatoire d'une loi uniforme.  Mais, par la 

suite (B4:E1053), seul le relai qui vient d'être remplacé est neuf et il faut modifier la formule 

pour tenir compte de l'âge des trois autres : par exemple, au moment de la 1
re

 panne (ligne 4), le 

premier relai fonctionne depuis 157,2 h (F3); comme sa durée de vie est de 166,0 h (B3), il lui 

reste 166,0 – 157,2 = 8,8 heures (B4) de survie.  De façon générale, pour calculer combien 

d'heures il reste à un relai donné, deux cas sont à considérer, selon que le relai vient d'être 

remplacé ou non, que nous illustrerons par deux exemples tirés de la ligne 4. 

 En E4, la formule commande d'abord de tester si E3 = $F3 : comme la réponse est positive, 

la fonction SI renvoie la 1
re

 valeur, qui représente la durée de vie du relai qui vient d'être 

remplacé. 

 En B4, la condition « B3 = $F3 » n'est pas satisfaite et Excel reportera dans la cellule la 

valeur du 2
e
 argument : par conséquent, B4 = B3 – $F3 = 182,3 – 162,1 = 20,2.  Autrement 

dit, le relai 1 fonctionnera encore pendant 20,2 heures. 

 

Les données de la colonne F représentent les délais (en heures) entre deux pannes consé-

cutives.  Comme la production de l'aciérie s'arrête dès que l'un des quatre relais tombe en 

panne, ces délais se calculent comme le minimum des durées de survie reportées dans les 

colonnes B à E. 

 

L'horloge de la colonne G donne l'écart (en heures) entre le début du processus et le moment où 

se produit une panne.  Par exemple, la 2
e
 (voir cellule G4) survient à l'instant  t = 162,1 + 4,8 = 

166,9.  En termes algébriques, le temps reporté dans la cellule G4 s'obtient comme la somme 

du temps à l'horloge sur la ligne précédente (G3) et du délai (F4) entre les pannes numéros 1 

et 2. 

 

Fichier : MOG8-18.xlsx Feuilles :  Pol1 et Fig1 

 
Cellule Formule Copiée dans :  

B3 =XMin+ALEA()*(XMax-XMin) C3:E3 

B4 =SI(B3=$F3;XMin+ALEA()*(XMax-XMin);B3-$F3) B4:E1053 

F3 =MIN(B3:E3) F4:F1053 

G3 =F3 --------------- 

G4 =G3+F4 G5:G1053 

 

 

Si l'entreprise décidait de remplacer les quatre relais après chaque panne, l'analyse précédente 

prendrait la forme illustrée à la figure 2 (voir page suivante).  L'approche est identique à celle 

de la figure 1, sauf que la formule de B4 n'apparaît pas cette fois et que celle de B3 est copiée 

dans la plage B3:E1002.  En effet, en vertu de la nouvelle politique, le laminoir est muni de 

quatre relais neufs après chaque panne, ce qui signifie que, sur toutes les lignes, le temps 

restant de fonctionnement de chacun coïncide avec sa durée de vie et obéit à une loi 

uniforme. 

Fichier : MOG8-18.xlsx Feuilles :  Pol2 et Fig2 

 
Cellule Formule Copiée dans :  

B3 =XMin+ALEA()*(XMax-XMin) B3:E1002 

F3 =MIN(B3:E3) F4:F1002 

G3 =F3 --------------- 

G4 =G3+F4 G5:G1002 

 

 

Les résultats des deux simulations précédentes sont résumés dans la figure 3 (voir ci-dessous).  

Les paramètres utilisés CoûtR = 850 et CoûtH = 1000 représentent, le premier, le coût d'un relai 

et, le second, le manque à gagner encouru par l'entreprise lorsque la production est arrêtée 

pendant une heure à cause d'une panne.  Dans les calculs associés à la politique 1 (colonne C), on 

ne tient pas compte des lignes 3 à 53 : en effet, au départ, les quatre relais sont neufs simultané-

ment, une situation exceptionnelle; on considère donc que la période couverte par les pannes 

numérotées de -50 à 0 constitue un temps de démarrage, à ignorer, et que, par la suite, le 

processus a atteint une allure de croisière qui est représentative. 

 

Fichier : MOG8-18.xlsx Feuilles :  Résultats et Fig3 

 
Cellule Formule Copiée dans :  

C19 ='Pol1'!G$1053-'Pol1'!G$53 --------------- 

D19 ='Pol2'!G$1002 --------------- 

C20 =C19/1000 D20 

C21 =(CoûtR+CoûtH)*1000/C19 --------------- 

D21 =(4*CoûtR+2*CoûtH)*1000/D19 --------------- 

 

Les coûts des politiques 1 et 2 sont exprimés en dollars par heure.  Dans les deux cas, ils se 

calculent en multipliant le coût associé à une panne par le nombre de pannes (1000), puis en 

divisant par la durée de la période (C19 ou D19).  Le 1
er

 facteur diffère selon la politique retenue. 

 Si l'entreprise remplace seulement le relai défectueux, une panne entraîne un déboursé de 

CoûtR = 850 dollars pour le relai neuf et un manque à gagner de CoûtH = 1000 dollars 

découlant de l'arrêt de la production pendant une heure; au total, CoûtR + CoûtH = 1850 

dollars. 
 

 Si la nouvelle politique est implantée, la production est interrompue pendant 2 heures, ce qui 

revient à (2 CoûtH) dollars, et il faut payer 4 relais neufs; au total, (2 CoûtR) + (4 CoûtH) = 

5400 dollars par panne. 

 

 

(a) La politique courante de gestion des pannes revient à 42,24$/h. 

 

(b) Si la nouvelle politique était implantée, la gestion des pannes reviendrait à 33,68$/h. 

 

(c) La nouvelle politique semble la plus intéressante : d'abord, et surtout, elle coûte bien 

moins cher; de plus, elle assure une production plus soutenue, plus régulière, ce qui facilite la 

planification. 


